LPduality.wxmx

TABLE OF CONTENTS

(%i4)
```
load(draw)$  set_draw_defaults(line_width=2, grid = [2,2], point_type = filled_circle,
         head_type = 'nofilled, head_angle = 20, head_length = 0.5,
           background_color = light_gray,  draw_realpart=false)$
 fpprintprec : 5$  ratprint : false$
```

(%i5)
```
load (simplex);
```

(%o5) *C:/maxima−5.43.2/share/maxima/5.43.2/share/simplex/simplex.mac*

(%i6)
```
load ("Econ1.mac");
```

(%o6) *c:/work5/Econ1.mac*

1 Preface

LPduality.wxmx assumes the reader has a general understanding of the simplex method of solving for optimal solutions of "linear programming" problems (LP problems). An introduction to simplex methods using Maxima can be found in our LPsimplex.wxmx.

1. We begin with a description of the tableau conventions in which the "z-row(s)" are on the bottom of the tableau (rather than the top, as we used in LPsimplex.wxmx).

2. We next discuss the systems of symmetric dual LP pairs, including the Duality Theorem, and the Complimentary Slackness Principle. Examples with and without artificial variables are solved, using Maxima. For LP's with Step 0 tableau forms which include artificial variables, we use B/N's "two phase simplex method", which splits the z-row into two rows. (See References) We show how the solution to the other dual pair LP is embedded in the final tableau values of the slack and surplus variables.

3. We next discuss systems of antisymmetric dual LP pairs. We show an explicit Maxima matrix calculation which allows a prediction of a solution of the other antisymmetric dual pair LP, given the Basis variables in the optimal tableau of one of the pair.

A code file Econ1.mac is available in the Economic Analysis with Maxima section of my CSULB webpage. This code file defines the Maxima functions used in this worksheet. Inside your worksheet, use load ("Econ1.mac");

This worksheet is one of a number of wxMaxima files available in the section
Economic Analysis with Maxima
on my CSULB webpage.

We have inserted this worksheet in the middle of Dowling's Ch. 13 worksheets, since it seems appropriate to discuss the solution of linear optimization (LP) problems before the solution of nonlinear optimization problems.

Edwin L. (Ted) Woollett
http://home.csulb.edu/~woollett/
Nov. 14, 2022

2 References

[B/N] Operations Research, 2nd ed, Schaum's Outlines, Richard Bronson and G. Naadimuthu

Sergiy Butenko, ISEN 620, A Survey of Optimization, 82 videos
https://www.youtube.com/playlist?list=PLY9yf2-4yyeQTLkCVFnGedyERuCjKW7kI

3 Tableau "z-row" on the Bottom Convention

The reference (B/N) Operations Research, 2nd ed, by Richard Bronson and G Naadimuthu, uses tableau conventions in which the z-row is the bottom row, instead of the top row (as was done in LPsimplex.wxmx). With minor changes, including the use of the new function bratio(RL,ncol) [b for bottom z-row] instead of tratio(RL,ncol) [t for top z-row], we can easily use the Bronson and Naadimuthu notation, which is also used in many other references. In particular, we can still use tableau(RL) and pivot1(RL, [nrow, ncol]). When using the two phase simplex method, in which the z-row is split into two rows on the bottom, we can use the function b2ratio(RL, ncol).

3.1 Step 0 Tableau Conventions

Let Xs be the matrix column vector of variables symbols, including slack, surplus and artificial variables.
Xo = matrix column vector of the Step 0 Basis variable symbols, in the order these variables appear in the constraints.
Co = matrix column vector of the coefficients of the step 0 basis variables in the objective, in the same order as the variable symbols in Xo.

Given the Step 0 LP: maximize z = Cs^t . Xs, such that As . Xs = E, with Xs >= 0, the step 0 maximization tableau is, using these matrices:

```
 |         Xs^t              |    rhs      | Basis
----------------------------------------------------------
 |          As               |  E          | Xo
----------------------------------------------------------
 |  Co^t . As - Cs^t         | Co^t . E  |   z
```

Given the Step 0 LP: minimize w = Cs^t . Xs, such that As . Xs = E, with Xs >= 0, the step 0 minimization tableau is, using these matrices:

```
 |         Xs^t              |     rhs       | Basis
----------------------------------------------------------
 |          As               |      E         |  Xo
----------------------------------------------------------
 |  Cs^t - Co^t . As         | - Co^t . E  |   z
```

3.2 B/N Prob. 3.1, Maximization, no Artificial Variables

maximize z = x1 + 9*x2 + x3,
  such that
   x1 + 2*x2 + 3*x3 <= 9,
 3*x1 + 2*x2 + 2*x3 <= 15,
x1,x2,x3 >= 0.

(%i7) maximize_lp(x1 + 9*x2 + x3, [x1 + 2*x2 + 3*x3 <= 9, 3*x1 + 2*x2 + 2*x3 <= 15],
[x1,x2,x3]);

(%o7) $[\frac{81}{2},[x3=0,x2=\frac{9}{2},x1=0]]$

We add nonnegative slack variables x4 and x5 to the right-hand sides of the condition inequalities, converting them into condition equalities, and arrive at the Step 0 form:

maximize z = x1 + 9*x2 + x3 + 0*x4 + 0*x5,
such that
  x1 + 2*x2 + 3*x3 + x4 = 9,
  3*x1 + 2*x2 + 2*x3 + x5 = 15,
with x1,x2,x3,x4,x5 >= 0.

An initial feasible solution has x1 = x2 = x3 = 0, x4 = 9, x5 = 15.

Let G^t mean the transpose of the matrix G in the following.

If the write the Step 0 LP using matrices,
  maximize z = C^t . X such that A . X = B, with X >=0,

for this problem we then have:

X^t = (x1, x2, x3, x4, x5), C^t = (1, 2, 3, 0, 0), A = matrix ([1,2,3,1,0], [3,2,2,0,1]),
B^t = (9, 15), Xo^t = (x4, x5) refers to the Basis variables in the step 0 tableau,
and Co^t = (0, 0) refers to the objective coefficients of the step 0 Basis variables Xo.
The quantity Co^t . B = 0 = initial value of z, and finally,
Co^t . A = (0, 0, 0, 0, 0) here.

The step 0 maximization tableau is, using these matrices,

```
   |        X^t          |   rhs      | Basis
---------------------------------------------------------
   |         A           |  B         | Xo
---------------------------------------------------------
   |  Co^t . A - C^t     | Co^t . B   |  z
```

in which Xo is the Step 0 Basis vector of variable symbols, in the order these basis variables appear in the step 0 constraint equations, with the same number of variables as the vector B. C is the "cost vector", and Co is the cost vector associated with the Step 0 basis vector Xo variables, and contains the coefficients of the Xo variables in the objective, in the same order as the variables in Xo.

The last row is the "z-row", with the middle section containing -C^t as the main piece. The first term, Co^t . A, is present just in case one or more of the basis vector Xo variables have nonzero objective coefficients. We want to start the simplex algorithm with all of the basis elements having zero z-row coefficients.

The list bL is a list of the Step 0 Basis variables, in the order they appear in the condition equations, with z added as the last element. In this example, r1 and r2 begin with the coefficients coming from A . X = B, and end with the elements of B.
In this example, the last row of the tableau (the z-row) is defined by the list r3.
The list vL begins with the variable symbols in the order they appear in X, and end with the words : rhs, Basis.

Step 0 tableau:

```
(%i14) vL : [x1,x2,x3,x4,x5,rhs,Basis]$
       bL : [x4,x5,z]$
       r1 : [1,2,3,1,0,9]$
       r2 : [3,2,2,0,1,15]$
       r3 : [-1,-9,-1,0,0,0]$
       RL : [r1,r2,r3]$
       tableau(RL);
```

(%o14)

$$\begin{pmatrix} x1 & x2 & x3 & x4 & x5 & rhs & Basis \\ 1 & 2 & 3 & 1 & 0 & 9 & x4 \\ 3 & 2 & 2 & 0 & 1 & 15 & x5 \\ -1 & -9 & -1 & 0 & 0 & 0 & z \end{pmatrix}$$

The most negative coefficient in the z-row is -9, so the pivot column is the x2 column, ncol = 2.

Because the z-row is now the last row, we cannot use tratio(RL, ncol). Instead, use the function bratio(RL, ncol) (b for bottom row is z-row) defined in Econ1.mac.

```
(%i15) bratio(RL, 2);
```

(%o15)

$$\begin{pmatrix} 4.5 & x4 \\ 7.5 & x5 \end{pmatrix}$$

The minimum ratio test is won by the x4 row (nrow = 1), so the pivot element has value 2, (intersection of row 1 and col 2). We can then use pivot1(RL, [nrow, ncol]) as usual for the first simplex step to get the Step 1 tableau:

(%i16) RL : pivot1 (RL, [1, 2])$

pivot row = 1 *pivot col =* 2 *value =* 2

x2 enters Basis, x4 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & x4 & x5 & rhs & Basis \\ \frac{1}{2} & 1 & \frac{3}{2} & \frac{1}{2} & 0 & \frac{9}{2} & x2 \\ 2 & 0 & -1 & -1 & 1 & 6 & x5 \\ \frac{7}{2} & 0 & \frac{25}{2} & \frac{9}{2} & 0 & \frac{81}{2} & z \end{pmatrix}$$

The z-row coefficients in the bottom row are now all positive, an optimum has been reached, with z* = 81/2, x1* = x3* = 0, x2* = 9/2, which is the same solution found by maximize_lp.

3.3 Minimization Example, no Artificial Variables

This minimization problem was considered in LPsimplex.wxmx in Sec. 5.2. The two slack variables were called s1 and s2 there. C^t stands for the transpose of C.

minimize w = -8*x1 - 10*x2 - 7*x3
 s.t. x1 + 3*x2 + 2*x3 <= 10,
 x1 + 5*x2 + x3 <= 8,
 with x1, x2, x3 >= 0.

```
(%i17) minimize_lp (-8*x1 - 10*x2 - 7*x3, [x1 + 3*x2 + 2*x3 <= 10,
           x1 + 5*x2 + x3 <= 8], [x1,x2,x3]);
(%o17) [-64,[x3=0,x2=0,x1=8]]
```

Converting from inequality conditions to equality conditions with slack variables x4, x5,

minimize w = C^t . X, s.t. A . X = B, with X>=0,
 which translates to:
minimize w = -8*x1 - 10*x2 - 7*x3 + 0*x4 + 0*x5,
s.t. x1 + 3*x2 + 2*x3 + x4 = 10,
 x1 + 5*x2 + x3 + x5 = 8,
 with x1, x2, x3, x4, x5 >= 0.

Here we have X^t = (x1, x2, x3, x4, x5), C^t = (-8, -10, -7, 0, 0), Xo^t = (x4,x5), Co^t = (0, 0), B^t = (10, 8), A = matrix([1,3,2,1,0], [1,5,1,0,1]).

the step 0 minimization tableau is, using these matrices:

```
|        X^t           |   rhs      | Basis
--------------------------------------------------
|         A            |   B        |  Xo
--------------------------------------------------
|  C^t - Co^t . A      | - Co^t . B |  z
```

in which Xo is the known basic feasible solution Basis vector, with the same number of elements as the rhs vector B, C is defined by min w = C^t . X, and Co is contains the objective coefficients of the Xo variables in the same order as in Xo. The last row is the "z-row", with the middle section containing C^t as the main piece; the second term (- Co^t . A) is present just in case one or more of the step 0 basis vector Xo elements have nonzero objective coefficients. We want to start the simplex algorithm with all of the basis elements having zero z-row coefficients.

The fact that we use C^t as the main part of the last row means that we are really going to maximize z = -w with the simplex algorithm.

The list r1 elements begine with the first row of A, the last element is the first element of B. Likewise r2 comes from A[2] and B[2,1]. r3 here is the bottom z-row, including the far right "current value of z": - Co^t . B.

```
(%i23) C : cvec([-8, -10, -7, 0, 0])$
       Co : cvec([0, 0])$
       B : cvec ([10,8])$
       A : matrix([1,3,2,1,0], [1,5,1,0,1])$
       "C^t - Co^t . A" = transpose(C) - transpose(Co) . A;
       "- Co^t . B" = - transpose(Co) . B;
```

(%o22) $C\hat{}t - Co\hat{}t\ .\ A = \begin{pmatrix} -8 & -10 & -7 & 0 & 0 \end{pmatrix}$

(%o23) $- Co\hat{}t\ .\ B = 0$

Step 0 tableau:

(%i30)
```
vL : [x1,x2,x3,x4,x5,rhs,Basis]$
bL : [x4,x5,z]$
r1 : endcons (B[1,1], A[1])$
r2 : endcons (B[2,1], A[2])$
r3 : [-8,-10,-7,0,0,0]$
RL : [r1,r2,r3]$
tableau(RL);
```

(%o30)

x1	x2	x3	x4	x5	rhs	Basis
1	3	2	1	0	10	x4
1	5	1	0	1	8	x5
−8	−10	−7	0	0	0	z

Choose the x2 column as the pivot column (col 2 here), use the minimum ratio test to pick the pivot row (and hence the "pivot element"). We can either do this "by hand",

(%i31)
```
[10/3, 8/5], numer;
```

(%o31) [3.3333, 1.6]

or use our function bratio(RL, ncol) [b for bottom z-row] to see the results of the minimum ratio test for column 2.

(%i32)
```
bratio(RL, 2);
```

(%o32)

3.3333	x4
1.6	x5

Step 1 tableau:

(%i33)
```
RL : pivot1 (RL, [2,2] )$
```

pivot row = 2 pivot col = 2 value = 5

x2 enters Basis, x5 leaves Basis

x1	x2	x3	x4	x5	rhs	Basis
$\frac{2}{5}$	0	$\frac{7}{5}$	1	$-\frac{3}{5}$	$\frac{26}{5}$	x4
$\frac{1}{5}$	1	$\frac{1}{5}$	0	$\frac{1}{5}$	$\frac{8}{5}$	x2
−6	0	−5	0	2	16	z

z has increased from 0 to 16 going from Step 0 to Step 1.

(%i34) bratio (RL, 1);

(%o34)
$$\begin{pmatrix} 13.0 & x4 \\ 8.0 & x2 \end{pmatrix}$$

Step 2 tableau:

(%i35) RL : pivot1(RL, [2,1])$

pivot row = 2 *pivot col* = 1 *value* = $\frac{1}{5}$

x1 enters Basis, x2 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & x4 & x5 & rhs & Basis \\ 0 & -2 & 1 & 1 & -1 & 2 & x4 \\ 1 & 5 & 1 & 0 & 1 & 8 & x1 \\ 0 & 30 & 1 & 0 & 8 & 64 & z \end{pmatrix}$$

With no negative coefficients in the last row, the optimum solution maximizing z = -w is z* = 64 with x1* = 8, x2* = x3* = 0. Hence min w = w* = -64, with x1* = 8, x2* = x3* = 0. Recall that our method for minimizing w is to maximize z = -w.

4 Symmetric Duals

We follow Ch. 4 in B/N.

Every linear program in the variables x1, x2, ..., xn has associated with it another linear program in the variables y1, y2, ..., ym (where m is the number of constraints in the original program), known as the "dual". The original program, called the "primal", completely determines the form of its dual.

For any matrix G, let G^t denote the transpose of G.

The dual of a (primal) linear program (LP) in the (nonstandard) matrix form is defined by:

Given: the primal LP in which X^t = (x1, x2, x3, ..., xn),
maximize z = C^t . X such that A . X <= B, with X >= 0, (1)

then the corresponding dual LP in, which Y^t = (y1,y2,y3,...,ym), is:
minimize w = B^t . Y such that A^t . Y >= C, with Y >= 0. (2)

Programs (1) and (2) are "symmetrical" in that both involve nonnegative variables and inequality constraints; they are known as the "symmetric duals" of each other. The dual variables y1, y2,..., ym are sometimes called "shadow costs".

Butenko's Video 27, LP Duality: motivation (Operations Research Video Series) presents a useful approach to thinking about the relation of primal to dual.

4.1 Duality Theorem

If an optimal solution exists to either the primal or symmetric dual program, then the other program also has an optimal solution and the two objective functions have the same optimal value.

In such situations, the optimal solution to the primal (dual) is found in the last row of the final simplex tableau for the dual (primal), in those columns associated with the slack or suplus variables (see Ex. 3). Since the solutions to both programs are obtained by solving either one, it may be computationally advantageous to solve a program's dual rather than the given program itself.

4.2 Example 1

Determine the symmetric dual of the program:

```
 maximize z = 20*x1 + 30*x2 + 40*x3 + 50*x4,
  subject to: 2*x1 + 6*x2 +   7*x3  +     x4 <= 5,
              3*x1 + 8*x2  +     x3   + 2*x4 <= 2,
                x1  + 5*x2  +  3*x3  +  4*x4 <= 1,
  with x1, x2, x3, x4 >= 0.
```

The dual LP is

```
 minimize w = 5*y1 + 2*y2 + y3,
 subject to: 2*y1 + 3*y2 +    y3 >= 20,
             6*y1 + 8*y2 + 5*y3 >= 30,
             7*y1 +    y2 + 3*y3 >= 40,
               y1 + 2*y2 + 4*y3 >= 50,
  with y1, y2, y3 >= 0.
```

The primal LP has three inequality constraint conditions, so the dual has three variables. The coefficients of the dual objective variables (y1,y2,y3) are the three right hand side constants of the three primal constraints. The primal LP has four variables (x1,x2,x3,x4) and hence the dual has four constraint conditions, with the right hand side constants (20, 30, 40, 50) equal to the primal objective coefficients.

Using matrices to create the symmetric dual:

(%i52)
```
print ("primal LP")$
X : cvec([x1,x2,x3,x4])$
C : cvec ([20, 30, 40, 50])$
print ("maximize")$
z = transpose (C) . X;
B : cvec ([5,2,1])$
A : matrix ([2,6,7,1], [3,8,1,2], [1,5,3,4])$
print ("such that")$
A . X <= B;
print ("with X >= 0")$
print ("corresponding dual")$
Y : cvec ([y1, y2, y3])$
print ("minimize")$
w = transpose (B) . Y;
print ("such that")$
transpose (A) . Y >= C;
print ("with Y >= 0")$
```

primal LP

maximize

(%o40) $z=50\ x4+40\ x3+30\ x2+20\ x1$

such that

(%o44) $$\begin{pmatrix} x4+7\ x3+6\ x2+2\ x1 \\ 2\ x4+x3+8\ x2+3\ x1 \\ 4\ x4+3\ x3+5\ x2+x1 \end{pmatrix} \leq \begin{pmatrix} 5 \\ 2 \\ 1 \end{pmatrix}$$

with X >= 0

corresponding dual

minimize

(%o49) $w=y3+2\ y2+5\ y1$

such that

(%o51) $$\begin{pmatrix} y3+3\ y2+2\ y1 \\ 5\ y3+8\ y2+6\ y1 \\ 3\ y3+y2+7\ y1 \\ 4\ y3+2\ y2+y1 \end{pmatrix} \geq \begin{pmatrix} 20 \\ 30 \\ 40 \\ 50 \end{pmatrix}$$

with Y >= 0

4.3 Example 2

Determine the symmetric dual of the program:
maximize z = 2*x1 + x2,
subject to: x1 + 5*x2 <= 10,
x1 + 3*x2 <= 6,
2*x1 + 2*x2 <= 8,
with x1, x2 >= 0.

The dual LP is
minimize w = 10*y1 + 6*y2 + 8*y3,
subject to: y1 + y2 + 2*y3 >= 2,
5*y1 + 3*y2 + 2*y3 >= 1,
with y1, y2, y3 >= 0.

```
(%i69) print ("primal LP")$
X : cvec([x1,x2])$
C : cvec ([2, 1])$
print ("maximize")$
z = transpose (C) . X;
B : cvec ([10,6,8])$
A : matrix ([1,5], [1,3], [2,2])$
print ("such that")$
A . X <= B;
print ("with X >= 0")$
print ("corresponding dual")$
Y : cvec ([y1, y2, y3])$
print ("minimize")$
w = transpose (B) . Y;
print ("such that")$
transpose (A) . Y >= C;
print ("with Y >= 0")$
```

primal LP

maximize

(%o57) $z=x2+2\,x1$

such that

(%o61) $\begin{pmatrix} 5\,x2+x1 \\ 3\,x2+x1 \\ 2\,x2+2\,x1 \end{pmatrix} \leq \begin{pmatrix} 10 \\ 6 \\ 8 \end{pmatrix}$

with X >= 0

corresponding dual

minimize

(%o66) $w=8\,y3+6\,y2+10\,y1$

such that

(%o68) $\begin{pmatrix} 2\,y3+y2+y1 \\ 2\,y3+3\,y2+5\,y1 \end{pmatrix} \geq \begin{pmatrix} 2 \\ 1 \end{pmatrix}$

with Y >= 0

4.4 Example 3, Solutions to Primal and Dual LP's of Ex. 2

Show that both the primal and dual programs in Example 2 have the same optimal value.

4.4.1 Primal LP Solution

The primal LP is:
maximize z = 2*x1 + x2,
subject to: x1 + 5*x2 <= 10,
x1 + 3*x2 <= 6,
2*x1 + 2*x2 <= 8,
with x1, x2 >= 0.

```
(%i70) maximize_lp (2*x1 + x2, [x1 + 5*x2 <= 10, x1 + 3*x2 <= 6,
 2*x1 + 2*x2  <= 8], [x1,x2]);
(%o70) [8,[x2=0,x1=4]]
```

Adding slack variables x3, x4, x5 to the primal LP constraints,
maximize z = 2*x1 + x2 + 0*x3 + 0*x4 + 0*x5,
subject to: x1 + 5*x2 + x3 = 10,
x1 + 3*x2 + x4 = 6,
2*x1 + 2*x2 + x5 = 8,
with x1, x2, x3, x4, x5 >= 0.

The initial feasible solution (ifs) is: x1 = x2 = 0, x3 = 10, x4 = 6, x5 = 8, z = 0.

Using matrix notation with C^t standing for the transpose of C,
maximize z = C^t . X, s.t. A . X = B, with X >=0,

The step 0 maximization tableau is, using these matrices,

```
|        X^t           |   rhs      | Basis
-------------------------------------------------------------
|         A            |  B         | Xo
-------------------------------------------------------------
|  Co^t . A - C^t      | Co^t . B |  z
```

C^t = (2,1,0,0,0), X^t = (x1,x2,x3,x4,x5),Xo^t = (x3,x4,x5), Co^t = (0,0,0),B^t = (10,6,8).
Xo = matrix column vector of the initial Basis variable symbols, in the order these variables appear in the condition equations.
Co = matrix column vector of the objective coefficients of the Xo variables, in the same order as the symbols in Xo.

An added step, if needed, is to calculate the two parts of the last row, using matrices.
Here it is obvious that Co^t . A = (0, 0, 0, 0, 0), Co^t . B = 0; the middle part of the z-row is -C^t = (-2,-1,0,0,0).

(%i76)
```
C : cvec([2,1,0,0,0])$
Co : cvec ([0,0,0])$
B : cvec ([10,6,8])$
A : matrix ([1,5,1,0,0], [1,3,0,1,0], [2,2,0,0,1])$
"Co^t . A - C^t" = transpose(Co) . A - transpose(C);
"Co^t . B" = transpose(Co) . B;
```

(%o75) $Co\hat{}t\ .\ A - C\hat{}t = \begin{pmatrix} -2 & -1 & 0 & 0 & 0 \end{pmatrix}$

(%o76) $Co\hat{}t\ .\ B = 0$

The list r1 comes from A[1] and B[1,1], likewise for list r2 and list r3, so each list contains 6 elements. Since we have gone through the trouble of defining A and B, we can show how this is done using endcons.

(%i77)
```
endcons ( B[1,1],  A[1]);
```

(%o77) [1,5,1,0,0,10]

The list r4 contains the elements of the last row, including the current z value Co^t . B as the last element.

The Step 0 tableau is (z-row on the bottom) [close to B/N's tableau but with Xo on the right-hand side instead of the left-hand side]:

(%i85)
```
vL : [x1,x2,x3,x4,x5,rhs,Basis]$
bL : [x3, x4, x5, z]$
r1 : endcons (B[1,1], A[1])$
r2 : endcons (B[2,1], A[2])$
r3 : endcons (B[3,1], A[3])$
r4 : [-2, -1, 0, 0, 0, 0]$
RL : [r1,r2,r3,r4]$
tableau(RL);
```

(%o85)

x1	x2	x3	x4	x5	rhs	Basis
1	5	1	0	0	10	x3
1	3	0	1	0	6	x4
2	2	0	0	1	8	x5
−2	−1	0	0	0	0	z

We choose the x1 column (col 1 here) as the pivot column since the most negative z-row coefficient (-2) is in col. 1. The ratios rhs/x1-coeff here are obvious, but we need practice with bratio(RL, ncol):

(%i86) bratio (RL, 1);

(%o86)
$$\begin{pmatrix} 10.0 & x3 \\ 6.0 & x4 \\ 4.0 & x5 \end{pmatrix}$$

Step 1 tableau using pivot1(RL, [nrow, ncol]):

(%i87) RL : pivot1 (RL, [3, 1])$

pivot row = 3 *pivot col =* 1 *value =* 2

x1 enters Basis, x5 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & x4 & x5 & rhs & Basis \\ 0 & 4 & 1 & 0 & -\frac{1}{2} & 6 & x3 \\ 0 & 2 & 0 & 1 & -\frac{1}{2} & 2 & x4 \\ 1 & 1 & 0 & 0 & \frac{1}{2} & 4 & x1 \\ 0 & 1 & 0 & 0 & 1 & 8 & z \end{pmatrix}$$

With no negative coefficients remaining in the z-row (the last row), z* = 8 with x1* = 4 and x2* = 0, in agreement with maximize_lp.

Recall that x3, x4, x5 primal variables are all slack.
In the final primal tableau,
**the first primal slack variable x3 has z-row coefficient = 0 which implies the
symmetric dual solution final tableau will have y1 = 0,
**the second primal slack variable x4 has z-row coefficient = 0 , which implies the
symmetric dual solution final tableau will have y2 = 0,
**the third primal slack variable x5 has z-row coefficient = 1, which implies the
symmetric dual solution final tableau will have y3 = 1.

Here is a matrix method of finding the solutions to the symmetric dual LP from the primal LP optimal tableau:

The basis variables in the optimal primal tableau are the first, third, and fourth variables of X, (x1,x3,x4).
Let Cb be the matrix column vector containing the first, third, and fourth coefficients in the primal vector C, and let Ab be the matrix containing the first, third, and fourth columns of the primal matrix A.
Then Y*^t = Cb^t . Ab^(-1) = transpose (Cb) . invert (Ab) = (y1, y2, y3).

(%i88) Cb : part (C, [1,3,4]);

(Cb) $\begin{pmatrix} 2 \\ 0 \\ 0 \end{pmatrix}$

(%i89) Ab : newM (A, [1,3,4]);

(Ab) $\begin{pmatrix} 1 & 1 & 0 \\ 1 & 0 & 1 \\ 2 & 0 & 0 \end{pmatrix}$

(%i90) transpose (Cb) . invert (Ab);

(%o90) $\begin{pmatrix} 0 & 0 & 1 \end{pmatrix}$

which are the values of the dual LP solution Y*^t = (y1, y2, y3).

4.4.2 Dual LP Solution with B/N's Two Phase Simplex Method

The dual LP is:
minimize w = 10*y1 + 6*y2 + 8*y3,
subject to: y1 + y2 + 2*y3 >= 2,
5*y1 + 3*y2 + 2*y3 >= 1,
with y1, y2, y3 >= 0.

(%i91) minimize_lp (10*y1 + 6*y2 + 8*y3, [y1 + y2 + 2*y3 >= 2, 5*y1 + 3*y2 + 2*y3 >= 1], [y1, y2, y3]);

(%o91) $[8,[y3=1,y2=0,y1=0]]$

So minimize_lp has found the same solution for y1,y2,y3 as found in the primal slack variable z-row coefficients in the optimal tableau.

To get constraint equations, subtract surplus (excess) variables y4 and y5 to get
minimize w = 10*y1 + 6*y2 + 8*y3 + 0*y4 + 0*y5,
subject to: y1 + y2 + 2*y3 - y4 = 2,
5*y1 + 3*y2 + 2*y3 - y5 = 1,
with y1, y2, y3, y4, y5 >= 0.

We cannot start with an initial feasible solution (ifs) y1 = y2 = y3 = 0, since that would require y4 = -2, y5 = -1, in contradiction with our nonnegativity assumptions for the variables. So add artificial variables y6 and y7 to the left hand sides of the constraint equations, and add M*y6 + M*y7 to w to get:
minimize w = 10*y1 + 6*y2 + 8*y3 + 0*y4 + 0*y5 + M*y6 + M*y7,
subject to: y1 + y2 + 2*y3 - y4 + y6 = 2,
5*y1 + 3*y2 + 2*y3 - y5 + y7 = 1,
with y1, y2, y3, y4, y5, y6, y7 >= 0.
Then our initial feasible solution is: y1 = y2 = y3 = y4 = y5 = 0, y6 = 2, y7 = 1.
We seek an optimum solution in which the values of the artificial variables y6 and y7 have both been driven to zero.

Given the LP:
minimize w = C^t . X, s.t. A . X = B, with X >= 0,

the step 0 minimization tableau is, using these matrices:

X^t	rhs	Basis
A	B	Xo
C^t - Co^t . A	- Co^t . B	z

Here, X^t = (y1,y2,y3,y4,y5,y6,y7), C^t = (10,6,8,0,0,M,M), B^t = (2,1),
Xo = matrix column vector of the initial Basis variable symbols, in the order these variables appear in the condition equations; Xo^t = (y6,y7),
Co = matrix column vector of the objective coefficients of the initial Basis variables, in the same order as the variable symbols in Xo; Co^t = (M,M).

The construction of this minimization tableau implies that the simplex algorithm is being used to find the maximum of z = -w.

```
(%i97) C : cvec ([10,6,8,0,0,M,M])$
       Co : cvec ([M,M])$
       B : cvec ([2, 1])$
       A : matrix ([1,1,2,-1,0, 1, 0],[5,3,2,0,-1, 0, 1])$
       "C^t - Co^t . A" = transpose(C) - transpose(Co) . A;
       "- Co^t . B" = - transpose(Co) . B;
```

(%o96) $C\hat{}t - Co\hat{}t\ .\ A = \begin{pmatrix} 10-6M & 6-4M & 8-4M & M & M & 0 & 0 \end{pmatrix}$

(%o97) $-Co\hat{}t\ .\ B = -3M$

Step 0 tableau; we split the z-row into two rows.
If the original z-row has the form
  [a1 + b1*M, a2 + b2*M,...]
we set r3 to the list [a1, a2, ...] and r4 to the list [b1, b2, ...].

As long as we have already gone to the trouble of defining A and B, we can define r1 and r2 using endcons.

Step 0 tableau, close to B/N's tableau, except Xo basis variables are on the rhs.

```
(%i105) vL : [y1, y2, y3, y4, y5, y6, y7, rhs, Basis]$
        bL : [y6, y7, z1, z2]$
        r1 : endcons (B[1,1], A[1])$
        r2 : endcons (B[2,1], A[2])$
        r3 : [10, 6, 8, 0, 0, 0, 0, 0]$
        r4 : [-6, -4, -4, 1, 1, 0, 0, -3]$
        RL : [r1, r2, r3, r4]$
        tableau(RL);
```

(%o105)

y1	y2	y3	y4	y5	y6	y7	rhs	Basis
1	1	2	−1	0	1	0	2	y6
5	3	2	0	−1	0	1	1	y7
10	6	8	0	0	0	0	0	z1
−6	−4	−4	1	1	0	0	−3	z2

Looking for the most negative coefficient in the bottom row, we choose the y1 column as the pivot column and use the minimum ratio test to choose the pivot row.

```
(%i106) [2/1, 1/5], numer;
(%o106) [2,0.2]
```

Since we have two z-rows, we need to use b2ratio (RL, ncol).

(%i107) b2ratio (RL, 1);

(%o107) $\begin{pmatrix} 2.0 & y6 \\ 0.2 & y7 \end{pmatrix}$

So row 2 is the pivot row.

Step 1 tableau:

(%i108) RL : pivot1(RL, [2, 1])$

pivot row = 2 *pivot col =* 1 *value =* 5
y1 enters Basis, y7 leaves Basis

$$\begin{pmatrix} y1 & y2 & y3 & y4 & y5 & y6 & y7 & rhs & Basis \\ 0 & \frac{2}{5} & \frac{8}{5} & -1 & \frac{1}{5} & 1 & -\frac{1}{5} & \frac{9}{5} & y6 \\ 1 & \frac{3}{5} & \frac{2}{5} & 0 & -\frac{1}{5} & 0 & \frac{1}{5} & \frac{1}{5} & y1 \\ 0 & 0 & 4 & 0 & 2 & 0 & -2 & -2 & z1 \\ 0 & -\frac{2}{5} & -\frac{8}{5} & 1 & -\frac{1}{5} & 0 & \frac{6}{5} & -\frac{9}{5} & z2 \end{pmatrix}$$

The most negative coefficient in the bottom row (excluding the "rhs" number) is -8/5 in column 3.

(%i109) b2ratio (RL, 3);

(%o109) $\begin{pmatrix} 1.125 & y6 \\ 0.5 & y1 \end{pmatrix}$

The minimum ratio test is won by row 2

Step 2 tableau:

(%i110) RL : pivot1 (RL, [2, 3])$

pivot row = 2 *pivot col =* 3 *value =* $\frac{2}{5}$

y3 enters Basis, y1 leaves Basis

$$\begin{pmatrix} y1 & y2 & y3 & y4 & y5 & y6 & y7 & rhs & Basis \\ -4 & -2 & 0 & -1 & 1 & 1 & -1 & 1 & y6 \\ \frac{5}{2} & \frac{3}{2} & 1 & 0 & -\frac{1}{2} & 0 & \frac{1}{2} & \frac{1}{2} & y3 \\ -10 & -6 & 0 & 0 & 4 & 0 & -4 & -4 & z1 \\ 4 & 2 & 0 & 1 & -1 & 0 & 2 & -1 & z2 \end{pmatrix}$$

The only negative coefficient in the last row (excluding the "rhs" number) is in col 5 and the only positive coefficient in col 5 (excluding the last two rows) is in row 1.

Step 3 tableau:

(%i111) RL : pivot1 (RL, [1, 5])$

pivot row = 1 *pivot col =* 5 *value =* 1

y5 enters Basis, y6 leaves Basis

$$\begin{pmatrix} y1 & y2 & y3 & y4 & y5 & y6 & y7 & rhs & Basis \\ -4 & -2 & 0 & -1 & 1 & 1 & -1 & 1 & y5 \\ \frac{1}{2} & \frac{1}{2} & 1 & -\frac{1}{2} & 0 & \frac{1}{2} & 0 & 1 & y3 \\ 6 & 2 & 0 & 4 & 0 & -4 & 0 & -8 & z1 \\ 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & z2 \end{pmatrix}$$

Artificial variables y6 and y7 have left the Basis and are hence equal to zero. We can delete columns 6 and 7, and then, since the bottom row will be all zeros, we can delete the last column.

(%i112) r1 : remL(RL[1], [6,7]);

(r1) $[-4,-2,0,-1,1,1]$

(%i113) r2 : remL(RL[2], [6,7]);

(r2) $[\frac{1}{2},\frac{1}{2},1,-\frac{1}{2},0,1]$

(%i114) r3 : remL(RL[3], [6,7]);

(r3) $[6,2,0,4,0,-8]$

(%i118)
```
vL : [y1,y2,y3,y4,y5,rhs,Basis]$
bL : [y5, y3, z]$
RL : [r1, r2, r3]$
tableau (RL);
```

(%o118)
$$\begin{pmatrix} y1 & y2 & y3 & y4 & y5 & rhs & Basis \\ -4 & -2 & 0 & -1 & 1 & 1 & y5 \\ \frac{1}{2} & \frac{1}{2} & 1 & -\frac{1}{2} & 0 & 1 & y3 \\ 6 & 2 & 0 & 4 & 0 & -8 & z \end{pmatrix}$$

Since there are no negative coefficients in the last row (excluding the "rhs" number), the maximum of z = -w is z* = -8 with y1* = y2* = 0, y3* = 1, which implies that the minimum of w = 10*y1 + 6*y2 + 8*y3 is w* = -z* = 8, with y1* = y2* = 0, y3* = 1, in agreement with minimize_lp.

Recall that y4 and y5 are surplus (excess) variables.

In the final symmetric dual tableau,
** the first dual surplus variable y4 has z-row coefficient equal to 4, implies primal x1 = 4.
**the second dual surplus variable y5 has z-row coefficient equal to 0,
implies primal x2 = 0.

Here is a matrix method of finding the solutions to the symmetric primal LP from the dual LP optimal tableau:
The basis variables in the dual optimal tableau are the third and fifth variables of Y (y3, y5).
Let Cb be the matrix column vector conaining the third and fifth coefficients in the dual LP vector C, and let Ab be the matrix containing the third and fifth columns of the dual LP matrix A.
Then X*^t = Cb^t . Ab^(-1) = transpose (Cb) . invert (Ab) gives the values of (x1, x2) in the primal LP optimal tableau.

(%i119)
```
Cb : part (C, [3,5]);
```

(Cb)
$$\begin{pmatrix} 8 \\ 0 \end{pmatrix}$$

(%i120)
```
Ab : newM (A, [3, 5]);
```

(Ab)
$$\begin{pmatrix} 2 & 0 \\ 2 & -1 \end{pmatrix}$$

(%i121) transpose (Cb) . invert (Ab);

(%o121) $\begin{pmatrix} 4 & 0 \end{pmatrix}$

If our goal is to find the optimal solution to the problem:

minimize w = 10*y1 + 6*y2 + 8*y3,
 subject to: y1 + y2 + 2*y3 >= 2,
                5*y1 + 3*y2 + 2*y3 >= 1,
 with y1, y2, y3 >= 0,

then, because of the need to introduce artificial variables, it is computationally advantageous to instead find the optimal solution to the dual problem:

maximize z = 2*x1 + x2,
 subject to: x1 + 5*x2 <= 10,
             x1 + 3*x2 <= 6,
             2*x1 + 2*x2 <= 8,
 with x1, x2 >= 0.

Note that the dual of the dual is the primal. (See, for example, Butenko's video 28.)

4.5 Complementary Slackness Principle

Given that the pair of symmetric duals have optimal solutions, then if the k'th constraint of one system holds as an inequality - i.e., the associated slack or surplus variable is positive for the optimal solution - the k'th component of the optimal solution of its symmetric dual is zero.

Applying this principle to Example 3, the first and second primal slack variables x3, x4 are both greater than zero, so the first and second symmetric dual variables y1 and y2 are equal to zero.

The second symmetric dual surplus variable y5 is greater than zero, so the second primal variable x2 is equal to zero.

4.5.1 Example 4

This example is taken from B/N Prob. 4.4.
We take the primal LP to be:

```
minimize w = x1 + x2 + x3 + x4 + x5 + x6,
  subject to
    x1 +                          x6    >= 7,
    x1 + x2                             >= 20,
         x2 +  x3                       >= 14,
               x3 +  x4                 >= 20,
                     x4 + x5            >= 10,
                          x5 +  x6       >= 5,
with all variables nonnegative.
```

We will find three alternative optimum solutions, each having the same minimum w value, with the first solution given by minimize_lp.

```
(%i122) minimize_lp (x1 + x2 + x3 + x4 + x5 + x6,
            [x1 + x6 >= 7, x1 + x2 >= 20,x2 +  x3 >= 14,x3 +  x4 >= 20,
                        x4 + x5 >= 10,x5 +  x6 >= 5],
            [x1,x2,x3,x4,x5,x6]);
(%o122) [45,[x5=0,x4=10,x3=10,x2=4,x6=5,x1=16]]
```

To solve this LP directly requirea the introduction of six surplus variables (x7,x8,...,x12), and then six artificial variables (x13,x14,...,x18), for a total of eighteen variables, and then employ the two phase simplex method.

In matrix form, minimize w = C^t . X such that A . X = B, X >= 0,
with X^t = (x1, x2,...,x18), C^t = (1,1,...,M) which has six 1's at the start, then six zeros, then six M's at the end. The matrix A has six rows, and B^t = (7,20,14,20,10,5).

4.5.2 Solving Ex 4 primal with 18 variables

(%i128)
```
A : matrix (
    [1,0,0,0,0,1,-1,0,0,0,0,0,1,0,0,0,0,0],
    [1,1,0,0,0,0,0,-1,0,0,0,0,0,1,0,0,0,0],
    [0,1,1,0,0,0,0,0,-1,0,0,0,0,0,1,0,0,0],
    [0,0,1,1,0,0,0,0,0,-1,0,0,0,0,0,1,0,0],
    [0,0,0,1,1,0,0,0,0,0,-1,0,0,0,0,0,1,0],
    [0,0,0,0,1,1,0,0,0,0,0,-1,0,0,0,0,0,1])$
B : cvec( [7,20,14,20,10,5])$
C : cvec ([1,1,1,1,1,1,0,0,0,0,0,0,M,M,M,M,M,M])$
Co : cvec ([M,M,M,M,M,M])$
"- Co^t . B " = - transpose (Co) . B;
" C^t - Co^t . A " = transpose(C) - transpose(Co) . A;
```

(%o127) $-Co\text{^}t\ .\ B = -76\,M$

(%o128) $C\text{^}t - Co\text{^}t\ .\ A =$

$$\begin{pmatrix} 1-2M & 1-2M & 1-2M & 1-2M & 1-2M & 1-2M & M & M & M & M & M & M & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

(%i140)
```
r7 : [1,1,1,1,1,1,0,0,0,0,0,0,0,0,0,0,0,0,0]$
r8 : [-2,-2,-2,-2,-2,-2,1,1,1,1,1,1,0,0,0,0,0,0,-76]$
r1 : endcons (B[1,1], A[1])$
r2 : endcons (B[2,1], A[2])$
r3 : endcons (B[3,1], A[3])$
r4 : endcons (B[4,1], A[4])$
r5 : endcons (B[5,1], A[5])$
r6 : endcons (B[6,1], A[6])$
RL : [r1,r2,r3,r4,r5,r6,r7,r8]$
vL : [x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,x11,x12,x13,x14,x15,x16,x17,x18,rhs,Basis]$
bL : [x13,x14,x15,x16,x17,x18,z1,z2]$
tableau (RL);
```

(%o140)

x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	x13	x14	x15	x16	x17	x18	rhs	Basis
1	0	0	0	0	1	−1	0	0	0	0	0	1	0	0	0	0	0	7	x13
1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	0	0	20	x14
0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	0	14	x15
0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	20	x16
0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	10	x17
0	0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	5	x18
1	1	1	1	1	1	0	0	0	0	0	0	0	0	0	0	0	0	0	z1
−2	−2	−2	−2	−2	−2	1	1	1	1	1	1	0	0	0	0	0	0	−76	z2

(%i141) b2ratio(RL, 1);

(%o141)

7.0	x13
20.0	x14
−	x15
−	x16
−	x17
−	x18

(%i142) RL : pivot1 (RL, [1,1])$

pivot row = 1 *pivot col =* 1 *value =* 1

x1 enters Basis, x13 leaves Basis

x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	x13	x14	x15	x16	x17	x18	rhs	Basis
1	0	0	0	0	1	−1	0	0	0	0	0	1	0	0	0	0	0	7	x1
0	1	0	0	0	−1	1	−1	0	0	0	0	−1	1	0	0	0	0	13	x14
0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	0	14	x15
0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	20	x16
0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	10	x17
0	0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	5	x18
0	1	1	1	1	0	1	0	0	0	0	0	−1	0	0	0	0	0	−7	z1
0	−2	−2	−2	−2	0	−1	1	1	1	1	1	2	0	0	0	0	0	−62	z2

(%i143) b2ratio(RL, 2);

(%o143)

−	x1
13.0	x14
14.0	x15
−	x16
−	x17
−	x18

(%i144) RL : pivot1(RL, [2,2])$

pivot row = 2 *pivot col =* 2 *value =* 1

x2 enters Basis, x14 leaves Basis

x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	x13	x14	x15	x16	x17	x18	rhs	Basis
1	0	0	0	0	1	−1	0	0	0	0	0	1	0	0	0	0	0	7	x1
0	1	0	0	0	−1	1	−1	0	0	0	0	−1	1	0	0	0	0	13	x2
0	0	1	0	0	1	−1	1	−1	0	0	0	1	−1	1	0	0	0	1	x15
0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	20	x16
0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	10	x17
0	0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	5	x18
0	0	1	1	1	1	0	1	0	0	0	0	0	−1	0	0	0	0	−20	z1
0	0	−2	−2	−2	−2	1	−1	1	1	1	1	0	2	0	0	0	0	−36	z2

(%i145) b2ratio(RL, 3);

(%o145)

−	x1
−	x2
1.0	x15
20.0	x16
−	x17
−	x18

(%i146) RL : pivot1(RL, [3, 3])$

pivot row = 3 *pivot col =* 3 *value =* 1

x3 enters Basis, x15 leaves Basis

x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	x13	x14	x15	x16	x17	x18	rhs	Basis
1	0	0	0	0	1	−1	0	0	0	0	0	1	0	0	0	0	0	7	x1
0	1	0	0	0	−1	1	−1	0	0	0	0	−1	1	0	0	0	0	13	x2
0	0	1	0	0	1	−1	1	−1	0	0	0	1	−1	1	0	0	0	1	x3
0	0	0	1	0	−1	1	−1	1	−1	0	0	−1	1	−1	1	0	0	19	x16
0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	10	x17
0	0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	5	x18
0	0	0	1	1	0	1	0	1	0	0	0	−1	0	−1	0	0	0	−21	z1
0	0	0	−2	−2	0	−1	1	−1	1	1	1	2	0	2	0	0	0	−34	z2

(%i147) b2ratio(RL, 4);

(%o147)

−	x1
−	x2
−	x3
19.0	x16
10.0	x17
−	x18

(%i148) RL : pivot1 (RL, [5, 4])$

pivot row = 5 *pivot col =* 4 *value =* 1

x4 enters Basis, x17 leaves Basis

x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	x13	x14	x15	x16	x17	x18	rhs	Basis
1	0	0	0	0	1	−1	0	0	0	0	0	1	0	0	0	0	0	7	x1
0	1	0	0	0	−1	1	−1	0	0	0	0	−1	1	0	0	0	0	13	x2
0	0	1	0	0	1	−1	1	−1	0	0	0	1	−1	1	0	0	0	1	x3
0	0	0	0	−1	−1	1	−1	1	−1	1	0	−1	1	−1	1	−1	0	9	x16
0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	10	x4
0	0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	5	x18
0	0	0	0	0	0	1	0	1	0	1	0	−1	0	−1	0	−1	0	−31	z1
0	0	0	0	0	0	−1	1	−1	1	−1	1	2	0	2	0	2	0	−14	z2

(%i149) b2ratio (RL, 7);

(%o149)

−	x1
13.0	x2
−	x3
9.0	x16
−	x4
−	x18

(%i150) RL : pivot1 (RL, [4, 7])$

pivot row = 4 *pivot col =* 7 *value =* 1

x7 enters Basis, x16 leaves Basis

x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	x13	x14	x15	x16	x17	x18	rhs	Basis
1	0	0	0	−1	0	0	−1	1	−1	1	0	0	1	−1	1	−1	0	16	x1
0	1	0	0	1	0	0	0	−1	1	−1	0	0	0	1	−1	1	0	4	x2
0	0	1	0	−1	0	0	0	0	−1	1	0	0	0	0	1	−1	0	10	x3
0	0	0	0	−1	−1	1	−1	1	−1	1	0	−1	1	−1	1	−1	0	9	x7
0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	10	x4
0	0	0	0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	5	x18
0	0	0	0	1	1	0	1	0	1	0	0	0	−1	0	−1	0	0	−40	z1
0	0	0	0	−1	−1	0	0	0	0	0	1	1	1	1	1	1	0	−5	z2

(%i151) b2ratio (RL, 5);

(%o151)

−	x1
4.0	x2
−	x3
−	x7
10.0	x4
5.0	x18

(%i152) RL : pivot1 (RL, [2, 5])$

pivot row = 2 *pivot col =* 5 *value =* 1

x5 enters Basis, x2 leaves Basis

x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	x13	x14	x15	x16	x17	x18	rhs	Basis
1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	0	0	20	x1
0	1	0	0	1	0	0	0	−1	1	−1	0	0	0	1	−1	1	0	4	x5
0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	0	14	x3
0	1	0	0	0	−1	1	−1	0	0	0	0	−1	1	0	0	0	0	13	x7
0	−1	0	1	0	0	0	0	1	−1	0	0	0	0	−1	1	0	0	6	x4
0	−1	0	0	0	1	0	0	1	−1	1	−1	0	0	−1	1	−1	1	1	x18
0	−1	0	0	0	1	0	1	1	0	1	0	0	−1	−1	0	−1	0	−44	z1
0	1	0	0	0	−1	0	0	−1	1	−1	1	1	1	2	0	2	0	−1	z2

(%i153) b2ratio (RL,6);

(%o153)

−	*x1*
−	*x5*
−	*x3*
−	*x7*
−	*x4*
1.0	*x18*

(%i154) RL : pivot1 (RL, [6,6])$

pivot row = 6 *pivot col =* 6 *value =* 1

x6 enters Basis, x18 leaves Basis

x1	*x2*	*x3*	*x4*	*x5*	*x6*	*x7*	*x8*	*x9*	*x10*	*x11*	*x12*	*x13*	*x14*	*x15*	*x16*	*x17*	*x18*	*rhs*	*Basis*
1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	0	0	20	*x1*
0	1	0	0	1	0	0	0	−1	1	−1	0	0	0	1	−1	1	0	4	*x5*
0	1	1	0	0	0	0	0	−1	0	0	0	0	0	1	0	0	0	14	*x3*
0	0	0	0	0	0	1	−1	1	−1	1	−1	−1	1	−1	1	−1	1	14	*x7*
0	−1	0	1	0	0	0	0	1	−1	0	0	0	0	−1	1	0	0	6	*x4*
0	−1	0	0	0	1	0	0	1	−1	1	−1	0	0	−1	1	−1	1	1	*x6*
0	0	0	0	0	0	0	1	0	1	0	1	0	−1	0	−1	0	−1	−45	*z1*
0	0	0	0	0	0	0	0	0	0	0	0	1	1	1	1	1	1	0	*z2*

All the artificial variables have left the Basis and hence equal zero. We can delete their columns of the tableau, and then delete the last row of zeros, otherwise preserving all other tableau elements.

(%i155) colL : [13,14,15,16,17,18]$

```
(%i166) r1 : remL (RL[1], colL)$
r2 : remL (RL[2], colL)$
r3 : remL (RL[3], colL)$
r4 : remL (RL[4], colL)$
r5 : remL (RL[5], colL)$
r6 : remL (RL[6], colL)$
r7 : remL (RL[7], colL)$
RL : [r1,r2,r3,r4,r5,r6,r7]$
vL : remL (vL, colL)$
bL : [x1,x5,x3,x7,x4,x6,z]$
tableau (RL);
```

(%o166)

x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	rhs	Basis
1	1	0	0	0	0	0	−1	0	0	0	0	20	x1
0	1	0	0	1	0	0	0	−1	1	−1	0	4	x5
0	1	1	0	0	0	0	0	−1	0	0	0	14	x3
0	0	0	0	0	0	1	−1	1	−1	1	−1	14	x7
0	−1	0	1	0	0	0	0	1	−1	0	0	6	x4
0	−1	0	0	0	1	0	0	1	−1	1	−1	1	x6
0	0	0	0	0	0	0	1	0	1	0	1	−45	z

Since the B/N tableau for minimization is actually maximizing z = -w, we have to take the minimum w to be +45, with x1 = 20, x2 = 0, x3 = 14, x4 = 6, x5 = 4, x6 = 1. This is a second optimum minimum for the primal, as compared with minimize_lp.

The z-row coefficients of the surplus variables x7, x8, ...,x12 gives the solution of the dual: y1 = 0, y2 = 1, y3 = 0, y4 = 1, y5 = 0, y6 = 1.

The only surplus variable with a positive value in the solution of the primal is x7 = 14, so the first constraint of the *primal* LP holds as an inequality, and the first decision variable of the *dual* y1 = 0, an example of the complementary slackness principle.

Here is a matrix method of finding the solutions to the symmetric dual LP from the primal LP optimal tableau:
The six basis variables in the optimal primal tableau are x1, x3, x4, x5, x6, x7.
Let Cb be the matrix column vector containing elements (1,2,4,5,6,7) of the primal vector C, and let Ab be the matrix containing columns (1,2,4,5,6,7) of the primal matrix A.
Then Y*^t = Cb^t . Ab^(-1) = transpose (Cb) . invert (Ab) = (y1, y2, y3, y4, y5, y6).

```
(%i167) Cb : part (C, [1,3,4,5,6,7]);
```

$$Cb = \begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 0 \end{pmatrix}$$

```
(%i168) Ab : newM (A, [1,3,4,5,6,7]);
```

$$Ab = \begin{pmatrix} 1 & 0 & 0 & 0 & 1 & -1 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 1 & 0 \end{pmatrix}$$

```
(%i169) transpose (Cb) . invert (Ab);
```

(%o169) $\begin{pmatrix} 0 & 1 & 0 & 1 & 0 & 1 \end{pmatrix}$

which is Y*^t = (y1,y2,y3,y4,y5,y6) for the optimal tableau of the symmetric dual LP.

4.5.3 Solving Ex 4 symmetric dual with 12 variables

Rather than solving the primal LP, it is simpler (if one wishes to avoid using maximize_lp) to solve the symmetric dual program:
maximize z = 7*y1 + 20*y2 + 14*y3 + 20*y4 + 10*y5 + 5*y6,
subject to
y1 + y2 <= 1,
y2 + y3 <= 1,
y3 + y4 <= 1,
y4 + y5 <= 1,
y5 + y6 <= 1,
y1 + y6 <= 1,
with all variables nonnegative.

```
(%i170) maximize_lp (7*y1 + 20*y2 + 14*y3 + 20*y4 + 10*y5 + 5*y6,
        [y1 + y2 <= 1,y2 + y3 <= 1,y3 + y4 <= 1,y4 + y5 <= 1,y5 + y6 <= 1, y1 + y6 <= 1],
        [y1,y2,y3,y4,y5,y6]);
```

(%o170) $[45,[y6=1,y5=0,y4=1,y3=0,y2=1,y1=0]]$

We introduce 6 slack variables, y7 through y12, on the left hand sides of the constraints.
The dual LP in matrix form is
maximize z = u^t . Y such that D . Y = v, with Y >= 0.

Given the Step 0 LP maximize z = u^t . Y, such that D . Y = v, with Y >= 0,
the step 0 maximization tableau is, using these matrices:

```
    |         Y^t              |    rhs      | Basis
-----------------------------------------------------------
    |          D               |  v          | Yo
-----------------------------------------------------------
    |    uo^t . D - u^t        | uo^t . v    |  z
```

(%i181)
```
Y : cvec ([y1,y2,y3,y4,y5,y6,y7,y8,y9,y10,y11,y12])$
D : matrix (
    [1,1,0,0,0,0,1,0,0,0,0,0],
    [0,1,1,0,0,0,0,1,0,0,0,0],
    [0,0,1,1,0,0,0,0,1,0,0,0],
    [0,0,0,1,1,0,0,0,0,1,0,0],
    [0,0,0,0,1,1,0,0,0,0,1,0],
    [1,0,0,0,0,1,0,0,0,0,0,1])$
v : cvec( [1,1,1,1,1,1])$
u : cvec ([7,20,14,20,10,5,0,0,0,0,0,0])$
"maximize  ";
z = transpose(u) . Y;
"such that ";
 D . Y = v;
uo : cvec ([0,0,0,0,0,0])$
" uo^t . v " =  transpose (uo) . v;
" uo^t . D - u^t " = transpose(uo) . D - transpose(u);
```

(%o175) *maximize*

(%o176) $z=5\,y6+10\,y5+20\,y4+14\,y3+20\,y2+7\,y1$

(%o177) *such that*

(%o178) $$\begin{pmatrix} y7+y2+y1 \\ y8+y3+y2 \\ y9+y4+y3 \\ y5+y4+y10 \\ y6+y5+y11 \\ y6+y12+y1 \end{pmatrix} = \begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \end{pmatrix}$$

(%o180) $uo\text{^}t\,.\,v\ =0$

(%o181) $uo\text{^}t\,.\,D-u\text{^}t\ =\begin{pmatrix} -7 & -20 & -14 & -20 & -10 & -5 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$

lme is our alias for list_matrix_entries (in Econ1.mac).

Step 0 tableau

```
(%i193) YL : lme (Y);
vL : flatten (endcons([rhs,Basis], YL));
bL : endcons (z, rest (YL,6));
r1 : endcons (v[1,1], D[1])$
r2 : endcons (v[2,1], D[2])$
r3 : endcons (v[3,1], D[3])$
r4 : endcons (v[4,1], D[4])$
r5 : endcons (v[5,1], D[5])$
r6 : endcons (v[6,1], D[6])$
r7 : endcons (0, lme (-transpose(u)))$
RL : [r1,r2,r3,r4,r5,r6,r7]$
tableau (RL);
```

(YL) [y1,y2,y3,y4,y5,y6,y7,y8,y9,y10,y11,y12]

(vL) [y1,y2,y3,y4,y5,y6,y7,y8,y9,y10,y11,y12,rhs,Basis]

(bL) [y7,y8,y9,y10,y11,y12,z]

(%o193)

y1	y2	y3	y4	y5	y6	y7	y8	y9	y10	y11	y12	rhs	Basis
1	1	0	0	0	0	1	0	0	0	0	0	1	y7
0	1	1	0	0	0	0	1	0	0	0	0	1	y8
0	0	1	1	0	0	0	0	1	0	0	0	1	y9
0	0	0	1	1	0	0	0	0	1	0	0	1	y10
0	0	0	0	1	1	0	0	0	0	1	0	1	y11
1	0	0	0	0	1	0	0	0	0	0	1	1	y12
−7	−20	−14	−20	−10	−5	0	0	0	0	0	0	0	z

```
(%i194) bratio (RL, 2);
```

(%o194)

1.0	y7
1.0	y8
−	y9
−	y10
−	y11
−	y12

Step 1 tableau:

(%i195) RL : pivot1 (RL, [1,2])$

pivot row = 1 *pivot col =* 2 *value =* 1

y2 enters Basis, y7 leaves Basis

y1	y2	y3	y4	y5	y6	y7	y8	y9	y10	y11	y12	rhs	Basis
1	1	0	0	0	0	1	0	0	0	0	0	1	y2
−1	0	1	0	0	0	−1	1	0	0	0	0	0	y8
0	0	1	1	0	0	0	0	1	0	0	0	1	y9
0	0	0	1	1	0	0	0	0	1	0	0	1	y10
0	0	0	0	1	1	0	0	0	0	1	0	1	y11
1	0	0	0	0	1	0	0	0	0	0	1	1	y12
13	0	−14	−20	−10	−5	20	0	0	0	0	0	20	z

(%i196) bratio (RL, 4);

(%o196)

−	y2
−	y8
1.0	y9
1.0	y10
−	y11
−	y12

Step 2 tableau:

(%i197) RL : pivot1 (RL, [3,4])$

pivot row = 3 *pivot col =* 4 *value =* 1

y4 enters Basis, y9 leaves Basis

y1	y2	y3	y4	y5	y6	y7	y8	y9	y10	y11	y12	rhs	Basis
1	1	0	0	0	0	1	0	0	0	0	0	1	y2
−1	0	1	0	0	0	−1	1	0	0	0	0	0	y8
0	0	1	1	0	0	0	0	1	0	0	0	1	y4
0	0	−1	0	1	0	0	0	−1	1	0	0	0	y10
0	0	0	0	1	1	0	0	0	0	1	0	1	y11
1	0	0	0	0	1	0	0	0	0	0	1	1	y12
13	0	6	0	−10	−5	20	0	20	0	0	0	40	z

(%i198) bratio (RL, 5);

(%o198)

−	y2
−	y8
−	y4
0.0	y10
1.0	y11
−	y12

Step 3 tableau:

(%i199) RL : pivot1 (RL, [4,5])$

pivot row = 4 *pivot col =* 5 *value =* 1

y5 enters Basis, y10 leaves Basis

y1	y2	y3	y4	y5	y6	y7	y8	y9	y10	y11	y12	rhs	Basis
1	1	0	0	0	0	1	0	0	0	0	0	1	y2
−1	0	1	0	0	0	−1	1	0	0	0	0	0	y8
0	0	1	1	0	0	0	0	1	0	0	0	1	y4
0	0	−1	0	1	0	0	0	−1	1	0	0	0	y5
0	0	1	0	0	1	0	0	1	−1	1	0	1	y11
1	0	0	0	0	1	0	0	0	0	0	1	1	y12
13	0	−4	0	0	−5	20	0	10	10	0	0	40	z

(%i200) bratio (RL, 6);

(%o200)

−	y2
−	y8
−	y4
−	y5
1.0	y11
1.0	y12

Step 4 tableau:

(%i201) RL : pivot1 (RL, [5,6])$

pivot row = 5 *pivot col* = 6 *value* = 1
y6 enters Basis, y11 leaves Basis

y1	y2	y3	y4	y5	y6	y7	y8	y9	y10	y11	y12	rhs	Basis
1	1	0	0	0	0	1	0	0	0	0	0	1	y2
−1	0	1	0	0	0	−1	1	0	0	0	0	0	y8
0	0	1	1	0	0	0	0	1	0	0	0	1	y4
0	0	−1	0	1	0	0	0	−1	1	0	0	0	y5
0	0	1	0	0	1	0	0	1	−1	1	0	1	y6
1	0	−1	0	0	0	0	0	−1	1	−1	1	0	y12
13	0	1	0	0	0	20	0	15	5	5	0	45	z

This is the optimized dual LP tableau with z* = 45, y1* = 0, y2* = 1, y3* = 0, y4* = 1, y5* = 0, and y6* = 1 as the solution to the dual, in agreement with the solution found by maximize_lp.

A third alternative solution to the *primal* LP is given by the z-row coefficients of the slack variables y7, y8,...,12. Namely, w* = 45 with x1* = 20, x2* = 0, x3* = 15, x4* = 5, x5* = 5, x6* = 0.

Using the fact that the basis variables in the optimal dual tableau are y2,y4,y5,y6,y8,y12, we can also use a matrix method to predict the same optimal solution for the primal variables as is given by the dual slack variable z-row coefficients.
The step 0 matrix dual LP description is: max z = u^t . Y such that D . Y = v, with Y>=0.

(%i203) BL : [2,4,5,6,8,12]$
ub : part (u, BL);

(ub)
$$\begin{pmatrix}20\\20\\10\\5\\0\\0\end{pmatrix}$$

(%i204) Db : newM (D, BL);

$$(Db)\quad \begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 1 \end{pmatrix}$$

(%i205) transpose (ub) . invert (Db);

$$(\%o205)\quad \begin{pmatrix} 20 & 0 & 15 & 5 & 5 & 0 \end{pmatrix}$$

which gives an optimal solution for the primal variables (x1,x2,x3,x4,x5,x6).

5 Unsymmetric Duals

Let C^t indicate the transpose of the matrix C, etc.

A. Given the primal LP:
maximize z = C^t . X such that A . X = B, X >= 0, (1)
the dual LP of (1) is:
minimize w = B^t . Y such that A^t . Y >= C, Y unrestricted in sign. (2)

Conversely, the dual LP of (2) is (1).

B. Given the primal LP:
minimize w = C^t . X such that A . X = B, X >= 0, (3)
the dual LP of (3) is:
maximize z = B^t . Y such that A^t . Y <= C, Y unrestricted in sign. (4)

Conversely, the dual of (4) is (3).

Since the dual of a program in "standard form" is not itself in "standard form", these duals are unsymmetric. Their forms are consistent with and a direct consequence of the definition of symmetric duals. (See B/N Problem 4.8)

5.1 Unsymmetric Duals Step 0 Notation

5.1.1 Dual Form (1) - (2) Step 0 Notation

The path taken to arrive at a suitable Step 0 LP problem often requires increasing the number of variables (by adding slack, surplus, and/or artificial variables). We then need to expand the sizes of matrices: X --> Xs, Y --> Ys, A --> As, etc.

Given the primal LP (1):
 maximize z = C^t . X such that A . X = B, X >= 0,
we write the Step 0 LP form as
 maximize z = Cs^t . Xs such that As . Xs = B, Xs >= 0. (1')

The dual LP of (1) is (2):
 minimize w = B^t . Y such that A^t . Y >= C, Y unrestricted in sign.
We write the Step 0 LP form as
 minimize w = u^t . Ys such that D . Ys = v, Ys >= 0. (2')

5.1.2 Dual Form (3) - (4) Step 0 Notation

Given the primal LP (3):
 minimize w = C^t . X such that A . X = B, X >= 0,
we write the Step 0 LP form as
 minimize w = Cs^t . Xs such that As . Xs = B, Xs >= 0, (3')

The dual LP of (3) is (4):
 maximize z = B^t . Y such that A^t . Y <= C, Y unrestricted in sign.
We write the Step 0 LP form as
 maximize z = u^t . Ys such that D . Ys = v, Ys >= 0. (4')

5.2 Example 5

Find the optimal solutions of the unsymmetric dual pair given and show that the solution of one of the pair will predict a solution of the other.

Given the primal LP
maximize z = x1 + 3*x2 - 2*x3 = C^t . X,
subject to 4*x1 + 8*x2 + 6*x3 = 25,
7*x1 + 5*x2 + 9*x3 = 30,
equivalent to A . X = B,
with x1,x2,x3 >= 0.

The (unsymmetric) dual LP is
minimize w = 25*y1 + 30*y2 = B^t . Y,
subject to 4*y1 + 7*y2 >= 1,
8*y1 + 5*y2 >= 3,
6*y1 + 9*y2 >= -2,
equivalent to A^t . Y >= C,
with y1 and y2 unrestricted in sign.

This pair is an example of the pair (1) & (2) above.

5.2.1 Primal LP Solution

The primal LP is:
maximize z = x1 + 3*x2 - 2*x3 = C^t . X,
subject to 4*x1 + 8*x2 + 6*x3 = 25,
7*x1 + 5*x2 + 9*x3 = 30,
equivalent to A . X = B,
with x1,x2,x3 >= 0.

```
(%i206) maximize_lp (x1 + 3*x2 - 2*x3, [4*x1 + 8*x2 + 6*x3 = 25, 7*x1 + 5*x2 + 9*x3 = 30],
        [x1,x2,x3]);
```

(%o206) $[\frac{70}{9},[x3=0,x2=\frac{55}{36},x1=\frac{115}{36}]]$

```
(%i207) float(%);
```

(%o207) $[7.7778,[x3=0.0,x2=1.5278,x1=3.1944]]$

To determine the optimum tableau with an initial feasible solution x1 = x2 = x3 = 0, we add artificial variables x4 and x5 to the left-hand side of the constraints so that the initial feasible solution is x1 = x2 = x3 = 0, x4 = 25 and x5 = 30. The Step 0 LP is then:

maximize z = x1 + 3*x2 - 2*x3 - M*x4 - M*x5 = Cs^t . Xs,
subject to 4*x1 + 8*x2 + 6*x3 + x4 = 25,
7*x1 + 5*x2 + 9*x3 + x5 = 30,
equivalent to As . Xs = B,
with x1,x2,x3,x4,x5 >= 0 or Xs >= 0.

5.2.2 Two Phase Simplex Method Primal LP Solution

We solve this Step 0 LP with artificial variables using the B/N "two phase simplex" method described in B/N, pp. 33,34.

Given the Step 0 LP:
maximize z = Cs^t . Xs, such that As . Xs = B, with Xs >= 0,

the step 0 maximization tableau is then, using these matrices:

```
|        Xs^t              |   rhs      | Basis
-----------------------------------------------------------
|         As               |    B       |  Xso
----------------------------------------------------------
| Cso^t . As - Cs^t        | Cso^t . B  |   z
```

Xso is the known initial feasible solution Basis vector of symbols.
Xso is defined using the Basis variable order in the constraint equations, and not necessarily with the order in Xs.
Xso has the same number of variables as the rhs vector B.
Cso is the vector of objective coefficients, taken from Cs, associated with the initial basis vector Xso, and in the same order as Xso.

The last row is the "z-row", with the middle section containing -Cs^t as the main piece. The first term, Cso^t . As, is present just in case one or more of the basis vector Xso variables have nonzero objective coefficients. We want to start the simplex algorithm with all of the step 0 basis variables having zero z-row coefficients.

The second term of the z-row is Cso^t . B, which is the step 0 value of z.
In our problem, Xs^t = (x1,x2,x3,x4,x5), Xso^t = (x4,x5).

```
(%i213) Cs : cvec([1,3,-2,-M,-M])$
        Cso : cvec ([-M, -M])$
        As : matrix( [4,8,6,1,0],[7,5,9,0,1] )$
        B : cvec ([25, 30])$
        "Cso^t . B" = transpose(Cso) . B;
        "Cso^t . As - Cs^t" = transpose(Cso) . As - transpose(Cs);
```

(%o212) $Cso\^{}t\ .\ B = -55\,M$

(%o213) $Cso\^{}t\ .\ As - Cs\^{}t = \begin{pmatrix} -11\,M-1 & -13\,M-3 & 2-15\,M & 0 & 0 \end{pmatrix}$

Step 0 tableau;
r1 and r2 are lists of the coefficients of the two Step 0 constraint equations, with 5 elements corresponding to the variables x1,x2,x3,x4,x5, and the sixth element corresponding to the right-hand side, coming from the vector B.
We split the z-row into two rows which supply the lists r3 and r4.
If the original z-row has the form (including the sixth element from the initial z value),
[a1 + b1*M, a2 + b2*M,...], then we set r3 to the list [a1, a2, ...] and
r4 to the list [b1, b2, ...].

For this problem,
Xs^t = (x1,x2,x3,x4,x5), Cs^t = (1,3,-2,-M,-M), z = Cs^t . Xs, Xso^t = (x4,x5),
Cso^t = (-M,-M), B^t = (25,30), As = matrix([4,8,6,1,0],[7,5,9,0,1]),
Cso^t . B = -55*M, Cso^t . As = (-11*M, -13*M, -15*M, -M, -M),
Cso^t . As - Cs^t = (-1 - 11*M, -3 - 13*M, 2 - 15*M, 0, 0)
This last (5 element) row is separated into two 5 element bottom rows:
(-1, -3, 2, 0, 0), with the 6'th element of r3 (rhs) = 0, the numerical part of from Cso^t . B,
(-11, -13, -15, 0, 0) with the 6'th element of r4 (rhs) = -55, from the symbolic part
of Cso^t . B).
We then apply the simplex method to four rows as a start, choosing the most negative coefficient in the bottom row (row 4 here) to define the "work column" or "pivot column".

Step 0 tableau: (Note that, in general, the order of the Step 0 Basic variables in the list bL follows the order those variables appear in Xso)

```
(%i221) vL : [x1,x2,x3,x4,x5,rhs,Basis]$
bL : [x4,x5,z1,z2]$
r1 : [4,8,6,1,0,25]$
r2 : [7,5,9,0,1,30]$
r3 : [-1,-3,2,0,0,0]$
r4 : [-11,-13,-15,0,0,-55]$
RL : [r1,r2,r3,r4]$
tableau(RL);
```

(%o221)

x1	x2	x3	x4	x5	rhs	Basis
4	8	6	1	0	25	x4
7	5	9	0	1	30	x5
−1	−3	2	0	0	0	z1
−11	−13	−15	0	0	−55	z2

Choose the x3 column as the pivot column, having the most negative coefficient in the bottom row (excluding the rhs number) . Use the minimum ratio test to select the pivot row, looking only at the r1 and r2 rows.

(%i222) b2ratio (RL, 3);

(%o222)
$$\begin{pmatrix} 4.1667 & x4 \\ 3.3333 & x5 \end{pmatrix}$$

Row 2 is the winning row.

Step 1 tableau:

(%i223) RL : pivot1(RL, [2, 3])$

pivot row = 2 *pivot col =* 3 *value =* 9

x3 enters Basis, x5 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & x4 & x5 & rhs & Basis \\ -\frac{2}{3} & \frac{14}{3} & 0 & 1 & -\frac{2}{3} & 5 & x4 \\ \frac{7}{9} & \frac{5}{9} & 1 & 0 & \frac{1}{9} & \frac{10}{3} & x3 \\ -\frac{23}{9} & -\frac{37}{9} & 0 & 0 & -\frac{2}{9} & -\frac{20}{3} & z1 \\ \frac{2}{3} & -\frac{14}{3} & 0 & 0 & \frac{5}{3} & -5 & z2 \end{pmatrix}$$

Column 2 is the pivot column,

(%i224) b2ratio(RL,2);

(%o224)
$$\begin{pmatrix} 1.0714 & x4 \\ 6.0 & x3 \end{pmatrix}$$

Row 1 is the winner of the minimum ratio test.

Step 2 tableau:

(%i225) RL : pivot1(RL, [1,2])$

$$pivot\ row = 1 \quad pivot\ col = 2 \quad value = \frac{14}{3}$$

x2 enters Basis, x4 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & x4 & x5 & rhs & Basis \\ -\frac{1}{7} & 1 & 0 & \frac{3}{14} & -\frac{1}{7} & \frac{15}{14} & x2 \\ \frac{6}{7} & 0 & 1 & -\frac{5}{42} & \frac{4}{21} & \frac{115}{42} & x3 \\ -\frac{22}{7} & 0 & 0 & \frac{37}{42} & -\frac{17}{21} & -\frac{95}{42} & z1 \\ 0 & 0 & 0 & 1 & 1 & 0 & z2 \end{pmatrix}$$

No more negative numbers in last row, x4 and x5 have left the Basis, so x4 = x5 = 0.
We can now remove columns 4 and 5. Then all numbers in last row equal 0, so we can remove the last row entirely.

The current value of the row 1 is given by RL[1].
We use our function remL(alist, item-numbers-list), which returns a depleted list missing those items in alist referred to by the integers in item-numbers-list.

(%i226) r1 : remL(RL[1], [4,5]);

(r1) $$[-\frac{1}{7},1,0,\frac{15}{14}]$$

(%i227) r2 : remL(RL[2], [4,5]);

(r2) $$[\frac{6}{7},0,1,\frac{115}{42}]$$

(%i228) r3 : remL (RL[3], [4,5]);

(r3) $$[-\frac{22}{7},0,0,-\frac{95}{42}]$$

Redefine the lists vL and bL, and RL is now a three element list, with elements r1,r2,and r3.

(%i232) vL : [x1,x2,x3,rhs,Basis]$
bL : [x2,x3,z]$
RL : [r1,r2,r3]$
tableau (RL);

(%o232)
$$\begin{pmatrix} x1 & x2 & x3 & rhs & Basis \\ -\frac{1}{7} & 1 & 0 & \frac{15}{14} & x2 \\ \frac{6}{7} & 0 & 1 & \frac{115}{42} & x3 \\ -\frac{22}{7} & 0 & 0 & -\frac{95}{42} & z \end{pmatrix}$$

(%i233) bratio (RL, 1);

(%o233)
$$\begin{pmatrix} - & x2 \\ 3.1944 & x3 \end{pmatrix}$$

Step 3 tableau:

(%i234) RL : pivot1(RL, [2, 1])$

pivot row = 2 *pivot col* = 1 *value* = $\frac{6}{7}$

x1 enters Basis, x3 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & rhs & Basis \\ 0 & 1 & \frac{1}{6} & \frac{55}{36} & x2 \\ 1 & 0 & \frac{7}{6} & \frac{115}{36} & x1 \\ 0 & 0 & \frac{11}{3} & \frac{70}{9} & z \end{pmatrix}$$

(%i235) float(RL);

(%o235) [[0.0,1.0,0.16667,1.5278],[1.0,0.0,1.1667,3.1944],[0.0,0.0,3.6667,7.7778]]

With no negative coefficients in the bottom row (the z-row), an optimum solution is z* = 70/9, x1* = 115/36, x2* = 55/36, x3* = 0 in agreement with maximize_lp.

Returning to the primal LP step 0 tableau with the knowledge that x1 and x2 are the Basis variables of the primal solution we can predict the solution of the dual LP from the properties of the primal initial step 0 tableau.

(%i236) display (Cs, As)$

$$Cs=\begin{pmatrix}1\\3\\-2\\-M\\-M\end{pmatrix}$$

$$As=\begin{pmatrix}4&8&6&1&0\\7&5&9&0&1\end{pmatrix}$$

The Basis variables in the optimal tableau are x1 and x2, elements 1 and 2 of Xs.
Let Cb be a column matrix vector whose elements are elements 1 and 2 of the vector Cs, and let Ab be a matrix made from columns 1 and 2 of the matrix As. Then the elements of the dual solution are given by
Cb^t . Ab^(-1) = transpose(Cb) . invert (Ab).

(%i238) Cb : part(Cs,[1,2]);
Ab : newM (As,[1,2]);

(Cb) $\begin{pmatrix}1\\3\end{pmatrix}$

(Ab) $\begin{pmatrix}4&8\\7&5\end{pmatrix}$

(%i239) transpose (Cb) . invert (Ab);

(%o239) $\begin{pmatrix}\frac{4}{9}&-\frac{1}{9}\end{pmatrix}$

This predicts that a solution of the (unsymmetric) dual LP is y1 = 4/9, y2 = -1/9.

5.2.3 Dual LP Solution

The (unsymmetric) dual is
minimize w = 25*y1 + 30*y2 = B^t . Y,
subject to 4*y1 + 7*y2 >= 1,
8*y1 + 5*y2 >= 3,
6*y1 + 9*y2 >= -2
equivalent to A^t . Y >= C,
with y1 and y2 unrestricted in sign.

(%i240) minimize_lp (25*y1 + 30*y2, [4*y1 + 7*y2 >= 1,
8*y1 + 5*y2 >= 3,
6*y1 + 9*y2 >= -2]);

(%o240) $[\frac{70}{9},[y2=-\frac{1}{9},y1=\frac{4}{9}]]$

The dual solution returned by minimize_lp is the same as was found using
Cb^t . Ab^(-1) on the primal lp result.

See B/N Prob. 2.6, 4.5, 4.7.
We need to take into account that y1 and y2 are unrestricted in sign before we can use the simplex method.
Set y1 = y3 - y4 and y2 = y5 - y6 with y3,y4,y5,y6 >= 0. Then multiply the last constraint by -1 to force a nonnegative right-hand side. This gives:

minimize w = 25*y3 - 25*y4 + 30*y5 - 30*y6,
subject to 4*y3 - 4*y4 + 7*y5 - 7*y6 >= 1,
8*y3 - 8*y4 + 5*y5 - 5*y6 >= 3,
-6*y3 + 6*y4 - 9*y5 + 9*y6 <= 2

Convert to standard form by subtracting surplus variables y7 and y8, respectively, from the left-hand sides of the first two constraints and add a slack variable y9 to the left-hand side of the third constraint. Then add artificial variables y10 and y11 respectively to the left-hand sides of the first two constraints. We finally arrive at:

minimize w = 25*y3 - 25*y4 + 30*y5 - 30*y6 +0*y7 + 0*y8 + 0*y9 + M*y10 + M*y11,

(equivalent to minimize w = u^t . Ys),

subject to: 4*y3 - 4*y4 + 7*y5 - 7*y6 - y7 + y10 = 1,
8*y3 - 8*y4 + 5*y5 - 5*y6 - y8 + y11 = 3,
-6*y3 + 6*y4 - 9*y5 + 9*y6 + y9 = 2,

(equivalent to D . Ys = v),

with y3,y4,y5,y6,y7,y8,y9,y10,y11 >= 0, (or Ys >= 0).

Then an initial feasible solution is, working top to bottom through the constraint equations:
y3 = y4 = y5 = y6 = y7 = y8 = 0, y10 = 1, y11 = 3, y9 = 2.

5.2.4 Two Phase Simplex Method Dual LP Solution

Using the Step 0 tableau matrix notation:

min w = u^t . Ys, such that D . Ys = v, with Ys >= 0, (u^t means transpose(u))

Ys^t = (y3, y4, y5, y6, y7, y8, y9, y10, y11),
u^t = (25, -25, 30, -30, 0, 0, 0, M, M), v^t = (1, 3, 2),
D = matrix ([4,-4,7,-7,-1,0,0,1,0], [8,-8,5,-5,0,-1,0,0,1], [-6,6,-9,9,0,0,1,0,0]),
** note that Yso is defined using the Basis variable's order in the constraint equations,
** not in the order found in Ys,
Yso^t = (y10, y11, y9),
uo^t = (M, M, 0) = transpose of the vector uo which contains the objective coefficients of the variables in Yso, in the same order as these variables appear in Yso (which is the order the basis variables appear in the constraint eqns).

The step 0 minimization tableau is, using these matrices:

```
   |        Ys^t           |  rhs       | Basis
---------------------------------------------------------
   |         D             |  v         |  Yso
---------------------------------------------------------
   |  u^t - uo^t . D       | - uo^t . v |    z
```

With artificial variables part of the Step 0 problem, we put the coefficients of M on the last line, splitting the z-row into two rows.

```
(%i246) u : cvec ([25,-25,30,-30,0,0, 0, M, M])$
uo : cvec ([M,M,0])$
v : cvec ([1, 3, 2])$
D : matrix ( [4,-4, 7,-7, -1,  0,  0,  1,  0],
             [8,-8, 5,-5,  0, -1,  0,  0, 1],
            [-6, 6, -9, 9, 0,  0,  1,  0, 0]  )$
"- uo^t . v" = - transpose(uo) . v;
"u^t - uo^t . D" = transpose(u) - transpose(uo) . D;
```

(%o245) $-uo\hat{}t\ .\ v = -4\,M$

(%o246) $u\hat{}t - uo\hat{}t\ .\ D = \begin{pmatrix} 25-12\,M & 12\,M-25 & 30-12\,M & 12\,M-30 & M & M & 0 & 0 & 0 \end{pmatrix}$

The last output is the z-row of the tableau (ignoring the current value of z given by - uo^t . v = -4*M), except we need to split the z-row into two rows since we have artificial variables. The z-row splits into two rows which supply the lists r4 and r5 in this problem.

If the original z-row has the form (including the sixth element from the initial z value), [a1 + b1*M, a2 + b2*M,...], then we set r4 to the list [a1, a2, ...] and r5 to the list [b1, b2, ...].

Since we have gone to the trouble of defining D and v, we can use these to define r1,r2,and r3.

Step 0 tableau: (Note that the order of the Step 0 Basic variables in the list bL follows the order those variables appear in Yso, ie., in the order of the Step 0 constraint equations.)

(%i255)
```
vL : [y3,y4,y5,y6,y7,y8,y9,y10,y11,rhs, Basis]$
bL : [y10,y11,y9, z1,z2]$
r1 : endcons (v[1,1], D[1])$
r2 : endcons (v[2,1], D[2])$
r3 : endcons (v[3,1], D[3])$
r4 : [25,-25,30,-30,0,0,0,0,0,0]$
r5 : [-12,12,-12,12,1,1,0,0,0,-4]$
RL : [r1,r2,r3,r4,r5]$
tableau (RL);
```

(%o255)

y3	y4	y5	y6	y7	y8	y9	y10	y11	rhs	Basis
4	−4	7	−7	−1	0	0	1	0	1	y10
8	−8	5	−5	0	−1	0	0	1	3	y11
−6	6	−9	9	0	0	1	0	0	2	y9
25	−25	30	−30	0	0	0	0	0	0	z1
−12	12	−12	12	1	1	0	0	0	−4	z2

Choose the y3 column (column 1) as the pivot column, y3 enters the Basis.

(%i256)
```
b2ratio (RL, 1);
```

(%o256)

0.25	y10
0.375	y11
−	y9

Row 1 wins the minimum ratio test.

Step 1 tableau:

(%i257) RL : pivot1 (RL, [1,1])$

pivot row = 1 *pivot col* = 1 *value* = 4

y3 enters Basis, y10 leaves Basis

$$\begin{pmatrix} y3 & y4 & y5 & y6 & y7 & y8 & y9 & y10 & y11 & rhs & Basis \\ 1 & -1 & \frac{7}{4} & -\frac{7}{4} & -\frac{1}{4} & 0 & 0 & \frac{1}{4} & 0 & \frac{1}{4} & y3 \\ 0 & 0 & -9 & 9 & 2 & -1 & 0 & -2 & 1 & 1 & y11 \\ 0 & 0 & \frac{3}{2} & -\frac{3}{2} & -\frac{3}{2} & 0 & 1 & \frac{3}{2} & 0 & \frac{7}{2} & y9 \\ 0 & 0 & -\frac{55}{4} & \frac{55}{4} & \frac{25}{4} & 0 & 0 & -\frac{25}{4} & 0 & -\frac{25}{4} & z1 \\ 0 & 0 & 9 & -9 & -2 & 1 & 0 & 3 & 0 & -1 & z2 \end{pmatrix}$$

Choose the y6 column (col 4) as the pivot column, y6 enters the basis. The only positive y6 coefficient is 9 in row 2, so y11 leaves the Basis.

(%i258) RL : pivot1 (RL, [2, 4])$

pivot row = 2 *pivot col* = 4 *value* = 9

y6 enters Basis, y11 leaves Basis

$$\begin{pmatrix} y3 & y4 & y5 & y6 & y7 & y8 & y9 & y10 & y11 & rhs & Basis \\ 1 & -1 & 0 & 0 & \frac{5}{36} & -\frac{7}{36} & 0 & -\frac{5}{36} & \frac{7}{36} & \frac{4}{9} & y3 \\ 0 & 0 & -1 & 1 & \frac{2}{9} & -\frac{1}{9} & 0 & -\frac{2}{9} & \frac{1}{9} & \frac{1}{9} & y6 \\ 0 & 0 & 0 & 0 & -\frac{7}{6} & -\frac{1}{6} & 1 & \frac{7}{6} & \frac{1}{6} & \frac{11}{3} & y9 \\ 0 & 0 & 0 & 0 & \frac{115}{36} & \frac{55}{36} & 0 & -\frac{115}{36} & -\frac{55}{36} & -\frac{70}{9} & z1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & z2 \end{pmatrix}$$

y10 and y11 have both left the Basis and are equal to 0, so we can delete the y10 and y11 columns, (columns 8 and 9) and then since we will have all zeros in the last row, we can delete the last row.

(%i259) r1 : remL (RL[1], [8, 9]);

(r1) $[1,-1,0,0,\frac{5}{36},-\frac{7}{36},0,\frac{4}{9}]$

(%i260) r2 : remL (RL[2], [8, 9]);

(r2) $[0,0,-1,1,\frac{2}{9},-\frac{1}{9},0,\frac{1}{9}]$

(%i261) r3 : remL (RL[3], [8, 9]);

(r3) $[0,0,0,0,-\frac{7}{6},-\frac{1}{6},1,\frac{11}{3}]$

(%i262) r4 : remL (RL[4], [8, 9]);

(r4) $[0,0,0,0,\frac{115}{36},\frac{55}{36},0,-\frac{70}{9}]$

We preserve everything else about the last tableau, including the order of the rows.
Note that bL needs to have the order top to botton of the last tableau basis variables.

```
(%i266) RL : [r1,r2,r3,r4]$
vL : [y3,y4,y5,y6,y7,y8,y9,rhs, Basis]$
bL : [y3, y6, y9, z]$
tableau (RL);
```

(%o266)

y3	y4	y5	y6	y7	y8	y9	rhs	Basis
1	−1	0	0	$\frac{5}{36}$	$-\frac{7}{36}$	0	$\frac{4}{9}$	y3
0	0	−1	1	$\frac{2}{9}$	$-\frac{1}{9}$	0	$\frac{1}{9}$	y6
0	0	0	0	$-\frac{7}{6}$	$-\frac{1}{6}$	1	$\frac{11}{3}$	y9
0	0	0	0	$\frac{115}{36}$	$\frac{55}{36}$	0	$-\frac{70}{9}$	z

There are no more negative z-row coefficients (ignoring the rhs number), so this tableau shows that z = -w has reached a maximum of -70/9, so w has reached a minimum value +70/9. Recall that the B/N tableau simplex method for min w actually maximizes z = -w.

Because the B/N tableau method finds min w by finding max z = -w, we have to take the negative of the last number in the bottom row as the minimum of w, so
w* = 70/9.

y1* = y3* - y4* = 4/9 - 0 = 4/9.
y2* = y5* - y6* = 0 - 1/9 = -1/9.

w* = 25*y1 + 30*y2 = 100/9 - 30/9 = 70/9.

(%i267) float(%);

(%o267)

y3	y4	y5	y6	y7	y8	y9	rhs	Basis
1.0	−1.0	0.0	0.0	0.13889	−0.19444	0.0	0.44444	y3
0.0	0.0	−1.0	1.0	0.22222	−0.11111	0.0	0.11111	y6
0.0	0.0	0.0	0.0	−1.1667	−0.16667	1.0	3.6667	y9
0.0	0.0	0.0	0.0	3.1944	1.5278	0.0	−7.7778	z

The Basis variables in the solution vector Ys* are y3, y6, and y9, elements 1,4,and 7 in Ys.

Let ub be a column matrix vector whose elements are elements 1, 4, and 7 of the vector u, and let Db be a matrix made from columns 1, 4 and 7 of the matrix D.
Then the elements of the primal solution are predicted to be:
ub^t . Db^(-1) = transpose(ub) . invert (Db)

```
(%i269) ub : part (u, [1,4,7]);
        Db : newM (D, [1,4,7]);
```

(ub) $\begin{pmatrix} 25 \\ -30 \\ 0 \end{pmatrix}$

(Db) $\begin{pmatrix} 4 & -7 & 0 \\ 8 & -5 & 0 \\ -6 & 9 & 1 \end{pmatrix}$

```
(%i270) transpose (ub) . invert (Db);
```

(%o270) $\begin{pmatrix} \frac{115}{36} & \frac{55}{36} & 0 \end{pmatrix}$

This predicts a solution of the primal should be x1 = 115/36, x2 = 55/36, x3 = 0, as we found in our primal solution.

5.3 Example 6

Determine the dual of the program
minimize z = 3*x1 + x2 + 0*x3 + 0*x4 + M*x5 + M*x6,
subject to: x1 + x2 - x3 + x5 = 7,
2*x1 + 3*x2 - x4 + x6 = 8,
with x1,x2,x3,x4,x5,x6 >= 0.
The (unsymmetric) dual is
maximize w = 7*y1 + 8*y2,
subject to y1 + 2*y2 <= 3,
y1 + 3*y2 <= 1,
- y1 <= 0,
-y2 <= 0,
y1 <= M,
y2 <= M,
with no further restrictions on the sign of y1 and y2.
Now -y1 <= 0 implies y1 >= 0, and -y2 <= 0 implies y2 >= 0, and y1 <= M simply requires y1 to be finite (a condition always presupposed), so we can simplify the dual to:
maximize w = 7*y1 + 8*y2,
subject to y1 + 2*y2 <= 3,
y1 + 3*y2 <= 1,
with y1, y2 >= 0.

5.4 Example 7

Find the optimal solutions of the unsymmetric dual pair given and show that the solution of one of the pair will predict a solution of the other.

Given the primal LP
minimize w = x1 + 2*x2 + x3 = C^t . X,
subject to x2 + x3 = 1,
3*x1 + x2 + 3*x3 = 4,
equivalent to A . X = B,
with x1,x2,x3 >= 0.

The (unsymmetric) dual LP is
maximize z = y1 + 4*y2 = B^t . Y,
subject to 3*y2 <= 1,
y1 + y2 <= 2,
y1 + 3*y2 <= 1,
equivalent to A^t . Y <= C,
with y1 and y2 unrestricted in sign.

This pair is an example of the unsymmetric dual pair (3) & (4) above.

5.4.1 Primal LP Solution

Given the primal LP
minimize w = x1 + 2*x2 + x3,
subject to x2 + x3 = 1,
3*x1 + x2 + 3*x3 = 4,
with x1,x2,x3 >= 0.

```
(%i271) minimize_lp (x1 + 2*x2 + x3, [x2 +  x3 = 1,
                         3*x1 + x2 + 3*x3 = 4],[x1,x2,x3]);
```

(%o271) $[\frac{4}{3},[x1=\frac{1}{3},x3=1,x2=0]]$

Add artificial variables x4 and x5 to the left-hand sides of the equality conditions.

minimize w = x1 + 2*x2 + x3 + M*x4 + M*x5 = Cs^t . Xs,
subject to
x2 + x3 + x4 = 1,
3*x1 + x2 + 3*x3 + x5 = 4,

equivalent to As . Xs = B, with Xs >= 0.

An initial feasible solution is then x1 = x2 = x3 = 0, x4 = 1, x5 = 4.

Let Xs^t = (x1,x2,x3,x4,x5), Xso^t = (x4,x5),
Cs^t = (1,2,1,M,M), Cso^t = (M,M), B^t = (1,4).

The step 0 minimization tableau is then, using these matrices:

```
 |        Xs^t              |  rhs          | Basis
------------------------------------------------------------
 |         As               |  B            |  Xso
------------------------------------------------------------
 |  Cs^t - Cso^t . As       | - Cso^t . B   |   z
```

```
(%i277) Cs : cvec ([1,2,1,M,M])$
        Cso : cvec ([M,M])$
        As : matrix ( [0,1,1,1,0], [3,1,3,0,1])$
        B : cvec ([1,4])$
        " - Cso^t . B " = - transpose (Cso) . B;
        " Cs^t - Cso^t . As " = transpose (Cs) - transpose (Cso) . As;
```

(%o276) $-Cso\hat{}t\ .\ B = -5\,M$

(%o277) $Cs\hat{}t - Cso\hat{}t\ .\ As = \begin{pmatrix} 1-3\,M & 2-2\,M & 1-4\,M & 0 & 0 \end{pmatrix}$

```
(%i285) r1 : endcons (B[1,1], As[1])$
        r2 : endcons (B[2,1], As[2])$
        r3 : [1,2,1,0,0,0]$
        r4 : [-3,-2,-4,0,0,-5]$
        RL : [r1,r2,r3,r4]$
        vL : [x1,x2,x3,x4,x5,rhs,Basis]$
        bL : [x4,x5,z1,z2]$
        tableau (RL);
```

(%o285)

x1	x2	x3	x4	x5	rhs	Basis
0	1	1	1	0	1	x4
3	1	3	0	1	4	x5
1	2	1	0	0	0	z1
−3	−2	−4	0	0	−5	z2

```
(%i286) b2ratio (RL, 3);
```

(%o286)

1.0	x4
1.3333	x5

```
(%i287) RL : pivot1 (RL, [1, 3])$
```

pivot row = 1 *pivot col =* 3 *value =* 1

x3 enters Basis, x4 leaves Basis

x1	x2	x3	x4	x5	rhs	Basis
0	1	1	1	0	1	x3
3	−2	0	−3	1	1	x5
1	1	0	−1	0	−1	z1
−3	2	0	4	0	−1	z2

```
(%i288) b2ratio (RL, 1);
```

(%o288)

−	x3
0.33333	x5

```
(%i289) RL : pivot1 (RL, [2, 1])$
```

pivot row = 2 *pivot col =* 1 *value =* 3

x1 enters Basis, x5 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & x4 & x5 & rhs & Basis \\ 0 & 1 & 1 & 1 & 0 & 1 & x3 \\ 1 & -\frac{2}{3} & 0 & -1 & \frac{1}{3} & \frac{1}{3} & x1 \\ 0 & \frac{5}{3} & 0 & 0 & -\frac{1}{3} & -\frac{4}{3} & z1 \\ 0 & 0 & 0 & 1 & 1 & 0 & z2 \end{pmatrix}$$

No more negative numbers in last row, x4 and x5 have left the Basis, so x4 = x5 = 0.
We can now remove columns 4 and 5. Then all numbers in last row equal 0, so
we can remove the last row entirely.

The current value of the row 1 is given by RL[1].
We use our function remL(alist, item-numbers-list), which returns a depleted list missing those items in alist referred to by the integers in item-numbers-list.

We retain the rest of the tableau. The list bL needs to be redefined to reflect the names of the Basis variables in this last tableau.

```
(%i290) r1 : remL (RL[1], [4,5]);
```

(r1) $[0,1,1,1]$

```
(%i291) r2 : remL (RL[2], [4,5]);
```

(r2) $[1,-\frac{2}{3},0,\frac{1}{3}]$

```
(%i292) r3 : remL (RL[3], [4,5]);
```

(r3) $[0,\frac{5}{3},0,-\frac{4}{3}]$

(%i296)
```
RL : [r1,r2,r3]$
vL : [x1,x2,x3,rhs, Basis]$
bL : [x3,x1,z]$
tableau (RL);
```

(%o296)
$$\begin{pmatrix} x1 & x2 & x3 & rhs & Basis \\ 0 & 1 & 1 & 1 & x3 \\ 1 & -\frac{2}{3} & 0 & \frac{1}{3} & x1 \\ 0 & \frac{5}{3} & 0 & -\frac{4}{3} & z \end{pmatrix}$$

The optimum primal solution has x1* = 1/3, x2* = 0, x3* = 1, min w = w* = 4/3, the negative of z. Our minimization tableau maximizes z = -w.

The Basis variables in the optimum tableau are x1 and x3, elements 1 and 3 of Xs.

Let Cb be a column vector with elements 1 and 3 from Cs, and Ab be the matrix with columns 1 and 3 chosen from As. We then calculate Cb^t . Ab^(-1) to get a prediction of values of y1 and y2 for the dual LP.

(%i297)
```
display (Cs, As)$
```

$$Cs=\begin{pmatrix} 1 \\ 2 \\ 1 \\ M \\ M \end{pmatrix}$$

$$As=\begin{pmatrix} 0 & 1 & 1 & 1 & 0 \\ 3 & 1 & 3 & 0 & 1 \end{pmatrix}$$

(%i299)
```
Cb : part(Cs,[1,3]);
Ab : newM (As,[1,3]);
```

(Cb)
$$\begin{pmatrix} 1 \\ 1 \end{pmatrix}$$

(Ab)
$$\begin{pmatrix} 0 & 1 \\ 3 & 3 \end{pmatrix}$$

(%i300)
```
transpose (Cb) . invert (Ab);
```

(%o300)
$$\begin{pmatrix} 0 & \frac{1}{3} \end{pmatrix}$$

Thus the solution to the primal LP, implies a solution to the dual LP is y1 = 0, y2 = 1/3.

What does maximize_lp give as a solution of the dual LP?

```
(%i301) maximize_lp (y1 + 4*y2,[3*y2 <= 1,y1 + y2 <= 2,y1 + 3*y2 <= 1]);
```

(%o301) $[\frac{4}{3},[y1=0,y2=\frac{1}{3}]]$

which agrees with the prediction.

5.4.2 Dual LP Solution

The (unsymmetric) dual LP is
  maximize z = y1 + 4*y2,
    subject to 3*y2 <= 1,
                y1 + y2 <= 2,
                y1 + 3*y2 <= 1,
              with y1 and y2 unrestricted in sign.

Let y1 = y3 - y4, and let y2 = y5 - y6.
Add slack variables y7,y8, y9 to the left-hand sides of the constraint inequalities to get constraint equations. We then have:

maximize z = y3 - y4 + 4*y5 - 4*y6 + 0*y7 + 0*y8 + 0*y9 = u^t . Ys,
subject to
   3*y5 - 3*y6 + y7 = 1,
  y3 - y4 + y5 - y6 + y8 = 2,
  y3 - y4 + 3*y5 - 3*y6 + y9 = 1,
  equivalent to D . Ys = v,
with y3,y4,y5,y6,y7,y8,y9 >= 0, or Ys >= 0.

The initial feasible solution is
y3 = y4 = y5 = y6 = 0, y7 = 1, y8 = 2, y9 = 1.
Ys^t = (y3,y4,y5,y6,y7,y8,y9), Yso^t = (y7,y8,y9), v^t = (1,2,1),
u^t = (1,-1,4,-4,0,0,0), uo^t = (0,0,0).

Given the LP:
maximize z = u^t . Ys, such that D . Ys = v, with Ys >= 0,
the step 0 maximization tableau is, using these matrices:

```
    |        Ys^t            | rhs      | Basis
-----------------------------------------------------------
    |         D              |  v       | Yso
-----------------------------------------------------------
    |  uo^t . D - u^t        | uo^t . v  |  z
```

```
(%i307) u : cvec ([1,-1,4,-4,0,0,0])$
v : cvec ([1,2,1])$
uo : cvec ([0,0,0])$
D : matrix ([0,0,3,-3,1,0,0], [1,-1,1,-1,0,1,0], [1,-1,3,-3,0,0,1])$
" uo^t . D - u^t " = transpose(uo) . D - transpose(u);
" uo^t . v " = transpose(uo) . v;
```

(%o306) $uo\text{^}t\ .\ D - u\text{^}t = \begin{pmatrix} -1 & 1 & -4 & 4 & 0 & 0 & 0 \end{pmatrix}$

(%o307) $uo\text{^}t\ .\ v = 0$

```
(%i315) r1 : endcons (v[1,1], D[1])$
r2 : endcons (v[2,1], D[2])$
r3 : endcons (v[3,1], D[3])$
r4 : [-1,1,-4,4,0,0,0,0]$
vL : [y3,y4,y5,y6,y7,y8,y9,rhs,Basis]$
bL : [y7,y8,y9,z]$
RL : [r1,r2,r3,r4]$
tableau (RL);
```

(%o315)

y3	y4	y5	y6	y7	y8	y9	rhs	Basis
0	0	3	−3	1	0	0	1	y7
1	−1	1	−1	0	1	0	2	y8
1	−1	3	−3	0	0	1	1	y9
−1	1	−4	4	0	0	0	0	z

```
(%i316) bratio (RL, 3);
```

(%o316)

0.33333	y7
2.0	y8
0.33333	y9

(%i317) RL : pivot1(RL, [1, 3])$

pivot row = 1 *pivot col =* 3 *value =* 3
y5 enters Basis, y7 leaves Basis

$$\begin{pmatrix} y3 & y4 & y5 & y6 & y7 & y8 & y9 & rhs & Basis \\ 0 & 0 & 1 & -1 & \frac{1}{3} & 0 & 0 & \frac{1}{3} & y5 \\ 1 & -1 & 0 & 0 & -\frac{1}{3} & 1 & 0 & \frac{5}{3} & y8 \\ 1 & -1 & 0 & 0 & -1 & 0 & 1 & 0 & y9 \\ -1 & 1 & 0 & 0 & \frac{4}{3} & 0 & 0 & \frac{4}{3} & z \end{pmatrix}$$

(%i318) bratio (RL,1);

(%o318)
$$\begin{pmatrix} - & y5 \\ 1.6667 & y8 \\ 0.0 & y9 \end{pmatrix}$$

(%i319) RL : pivot1 (RL, [3, 1])$

pivot row = 3 *pivot col =* 1 *value =* 1
y3 enters Basis, y9 leaves Basis

$$\begin{pmatrix} y3 & y4 & y5 & y6 & y7 & y8 & y9 & rhs & Basis \\ 0 & 0 & 1 & -1 & \frac{1}{3} & 0 & 0 & \frac{1}{3} & y5 \\ 0 & 0 & 0 & 0 & \frac{2}{3} & 1 & -1 & \frac{5}{3} & y8 \\ 1 & -1 & 0 & 0 & -1 & 0 & 1 & 0 & y3 \\ 0 & 0 & 0 & 0 & \frac{1}{3} & 0 & 1 & \frac{4}{3} & z \end{pmatrix}$$

With no more negative coefficients in the bottom z-row, the optimum found is
z* = 4/3, with y3* = y4* =y6* = y7* = y9* = 0, y5* = 1/3, y8* = 5/3.
y1* = y3* - y4* = 0, y2* = y5* - y6* = 1/3, which agrees with the solution found by maximize_lp.

The Basis variables in the optimal tableau are y3, y5, and y8, respectively elements 1, 3, and 6 in Ys.
Let ub be a matrix column vector made up of elements 1, 3 and 6 in u,
Let Db be a matrix made up of columns 1, 3 and 6 of D.
Then a predicted solution of the primal is ub^t . Db^(-1) = transpose(ub) . invert(Db).

(%i320) display (u, D)$

$$u=\begin{pmatrix}1\\-1\\4\\-4\\0\\0\\0\end{pmatrix}$$

$$D=\begin{pmatrix}0 & 0 & 3 & -3 & 1 & 0 & 0\\1 & -1 & 1 & -1 & 0 & 1 & 0\\1 & -1 & 3 & -3 & 0 & 0 & 1\end{pmatrix}$$

(%i322) ub : part (u, [1, 3, 6]);
Db : newM (D, [1, 3, 6]);

(ub) $\begin{pmatrix}1\\4\\0\end{pmatrix}$

(Db) $\begin{pmatrix}0 & 3 & 0\\1 & 1 & 1\\1 & 3 & 0\end{pmatrix}$

(%i323) transpose (ub) . invert (Db);

(%o323) $\begin{pmatrix}\frac{1}{3} & 0 & 1\end{pmatrix}$

A predicted solution of the primal is then x1 = 1/3, x2 = 0, x3 = 1, which agrees with our simplex solution of the primal LP.

6 General Rules For Forming Duals

Butenko's Video 28 (Dual of the Dual LP and Rules for Forming the Dual) considers general rules for forming duals, with the table:

```
(%i324) matrix (["Primal (max )", |,"Dual (min)"], ["==========",|, "=========="],
    ["(=) Constraint", |,"Unrestricted Sign Var."],["--------------",|, "--------------"],
   [" (<=) Constraint",|, "Non-negative Var."], ["--------------", |,"--------------"],
   ["Unrestricted Sign Var.", |," (=) Constraint"],["--------------",|, "--------------"],
   ["Non-negative Var.",|, " (>=) Constraint"] );
```

(%o324)

Primal (max)	|	*Dual (min)*
==========	|	==========
(=) Constraint	|	*Unrestricted Sign Var.*
--------------	|	--------------
(<=) Constraint	|	*Non−negative Var.*
--------------	|	--------------
Unrestricted Sign Var.	|	*(=) Constraint*
--------------	|	--------------
Non−negative Var.	|	*(>=) Constraint*

In using this table, we associate the first dual variable y1 with the first constraint, the dual variable y2 with the second constraint, etc. Thus if the second primal constraint condition was an equality (=) constraint, y2 would be unrestricted in sign.

6.1 Example 8

Given the primal LP
maximize z = 2*x1 + x2
  such that
    x1 + x2 = 2,
    2*x1 - x2 >= 3,
   x1 - x2 <= 1,
  with x1 >= 0, and x2 unrestricted (-urs).

A first step is to put the primal LP in "canonical form" for maximization, in which all inequality constraints have the form (<=). We then need to multiply both sides of the second constraint by (-1) which will reverse the sense of the inequality.

maximize z = 2*x1 + x2
  such that
    x1 + x2 = 2,
    -2*x1 + x2 <= -3,
   x1 - x2 <= 1,
  with x1 >= 0, and x2 unrestricted (-urs).

Because the first primal constraint is an equality constraint, the first dual variable y1 will be unrestricted in sign. The second and third dual variables (y2,y3) will be non-negative variables since the second and third primal constraints are (<=) inequality constraints.

Since x1 >= 0 in the primal, the dual constraint condition associated with the x1 coefficients will be a (>=) type of inequality constraint for minimization. Since x2 is -urs, the dual constraint condition associated with the x2 coefficients will be an equality (=) type of condition.

We then have the dual LP:

minimize w = 2*y1 - 3*y2 + y3,
such that
y1 - 2*y2 + y3 >= 2,
y1 + y2 - y3 = 1,
with y1 unrestricted, and y2,y3 >= 0.

6.2 Example 9

Assume the primal LP is:

minimize w = 0.4*x1 + 0.5*x2,
subject to
0.3*x1 + 0.1*x2 <= 2.7,
0.5*x1 + 0.5*x2 = 6,
0.6*x1 + 0.4*x2 >= 6,
with x1, x2 >= 0.

6.2.1 Finding the Dual LP

To find the dual LP, we first put the primal LP in canonical form for minimization, with (>=) type constraints (or equality) by multiplying the first constraint condition by (-1) which will change the sense of the inequality condition.

minimize w = 0.4*x1 + 0.5*x2,
subject to
- 0.3*x1 - 0.1*x2 >= - 2.7,
0.5*x1 + 0.5*x2 = 6,
0.6*x1 + 0.4*x2 >= 6,
with x1, x2 >= 0.

To find the dual LP we associate y1 with the first constraint of the canonical form, y2 with the second, y3 with the third. Since the second constraint is an equality condition, y2 will be unrestricted in sign (urs). Since the first and third constraints for minimization are (>=) type, y1 and y3 will be non-negative variables.

Since both x1 and x2 are non-negative variables, the dual constraint conditions arising from both the x1 and x2 coefficients will be of type (<=).

We then have for the dual LP:
maximize z = -2.7*y1 + 6*y2 + 6*y3,
subject to
-0.3*y1 + 0.5*y2 + 0.6*y3 <= 0.4,
-0.1*y1 + 0.5*y2 + 0.4*y3 <= 0.5,
with y1, y3 >= 0, and y2 unrestricted in sign (urs).

6.2.2 Simplex Tableau Solution to Primal LP

The primal LP is:
minimize w = 0.4*x1 + 0.5*x2,
subject to
0.3*x1 + 0.1*x2 <= 2.7,
0.5*x1 + 0.5*x2 = 6,
0.6*x1 + 0.4*x2 >= 6,
with x1, x2 >= 0.

```
(%i325) minimize_lp (0.4*x1 + 0.5*x2, [0.3*x1 + 0.1*x2 <= 2.7,
                                      0.5*x1 + 0.5*x2 = 6,
                                      0.6*x1 + 0.4*x2 >= 6], [x1,x2]);
```

Warning: linear_program(A,b,c): non-rat inputs found, epsilon_lp=1.e-8.
Warning: Solution may be incorrect.

(%o325) [5.25,[x2=4.5,x1=7.5]]

Convert to Step 0 form to apply the simplex method.
Add slack variable x3 to left-hand side of first constraint, artificial variable x4 to left-hand side of the second constraint, and subtract surplus variable x5 from the lhs of 3'rd constraint, add artificial variable x6 to lhs 3'rd constraint, to have three constraint equations, add M*x4 + M*x6 penalty terms to the rhs of the objective.

minimize w = 0.4*x1 + 0.5*x2 + 0*x3 + M*x4 + 0*x5 + M*x6,
 equivalent to min w = Cs^t . Xs,
 subject to
  0.3*x1 + 0.1*x2 + x3 = 2.7,
  0.5*x1 + 0.5*x2 + x4 = 6,
  0.6*x1 + 0.4*x2 - x5 + x6 = 6,
equivalent to As . Xs = B,
with x1, x2, x3, x4, x5, x6 >= 0, or Xs >= 0.

An initial (Step 0) feasible solution is
x1 = x2 = x5 = 0, x3 = 2.7, x4 = 6, x6 = 6.
Xs^t = (x1,x2,x3,x4,x5,x6), Xo^t = (x3, x4, x6),
Cs^t = (0.4, 0.5, 0, M, 0, M), Cso^t = (0, M, M)

Xo = matrix column vector of the Step 0 Basis variable symbols, in the order these variables appear in the constraint equations.
Cso = matrix column vector of the objective coefficients of the initial Basis variables Xo, in the same order as the symbols in Xo.
Given the Step 0 LP minimize w = Cs^t . Xs, such that As . Xs = B, with Xs >= 0, the step 0 minimization tableau is, using these matrices:

```
|        Xs^t            |   rhs         | Basis
-----------------------------------------------------------------
|         As             |   B           |  Xo
-----------------------------------------------------------------
|  Cs^t - Cso^t . As     | - Cso^t . B   |  z
```

```
(%i331) Cs : cvec([0.4, 0.5, 0, M, 0, M])$
        Cso : cvec ([0, M, M])$
        As : matrix( [0.3, 0.1, 1, 0, 0, 0],[0.5, 0.5, 0, 1, 0, 0], [0.6, 0.4, 0, 0, -1, 1] )$
        B : cvec ([2.7, 6, 6])$
        "-Cso^t . B" = - transpose(Cso) . B;
        " Cs^t - Cso^t . As " = transpose(Cs) - transpose(Cso) . As;
```

(%o330) $-Cso\text{^}t\ .\ B=-12\ M$

(%o331) $Cs\text{^}t-Cso\text{^}t\ .\ As=\begin{pmatrix}0.4-1.1\ M & 0.5-0.9\ M & 0 & 0 & M & 0\end{pmatrix}$

```
(%i340) vL : [x1, x2, x3, x4, x5, x6, rhs, Basis]$
bL : [x3, x4, x6, z1, z2]$
r1 : endcons (B[1,1], As[1])$
r2 : endcons (B[2,1], As[2])$
r3 : endcons (B[3,1], As[3])$
r4 : [0.4, 0.5, 0, 0, 0, 0, 0]$
r5 : [-1.1, -0.9, 0, 0, 1, 0, -12]$
RL : [r1,r2,r3,r4,r5]$
tableau (RL);
```

(%o340)

x1	x2	x3	x4	x5	x6	rhs	Basis
0.3	0.1	1	0	0	0	2.7	x3
0.5	0.5	0	1	0	0	6	x4
0.6	0.4	0	0	−1	1	6	x6
0.4	0.5	0	0	0	0	0	z1
−1.1	−0.9	0	0	1	0	−12	z2

```
(%i341) b2ratio (RL, 1);
```

(%o341)

9.0	x3
12.0	x4
10.0	x6

```
(%i342) RL : pivot1 (RL, [1, 1])$
```

pivot row = 1 *pivot col =* 1 *value =* 0.3

x1 enters Basis, x3 leaves Basis

x1	x2	x3	x4	x5	x6	rhs	Basis
1	$\frac{1}{3}$	$\frac{10}{3}$	0	0	0	9	x1
0	$\frac{1}{3}$	$-\frac{5}{3}$	1	0	0	$\frac{3}{2}$	x4
0	$\frac{1}{5}$	−2	0	−1	1	$\frac{3}{5}$	x6
0	$\frac{11}{30}$	$-\frac{4}{3}$	0	0	0	$-\frac{18}{5}$	z1
0	$-\frac{8}{15}$	$\frac{11}{3}$	0	1	0	$-\frac{21}{10}$	z2

(%i343) b2ratio (RL, 2);

(%o343)
$$\begin{pmatrix} 27.0 & x1 \\ 4.5 & x4 \\ 3.0 & x6 \end{pmatrix}$$

(%i344) RL : pivot1 (RL, [3, 2])$

$$pivot\ row = 3 \quad pivot\ col = 2 \quad value = \frac{1}{5}$$

x2 enters Basis, x6 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & x4 & x5 & x6 & rhs & Basis \\ 1 & 0 & \frac{20}{3} & 0 & \frac{5}{3} & -\frac{5}{3} & 8 & x1 \\ 0 & 0 & \frac{5}{3} & 1 & \frac{5}{3} & -\frac{5}{3} & \frac{1}{2} & x4 \\ 0 & 1 & -10 & 0 & -5 & 5 & 3 & x2 \\ 0 & 0 & \frac{7}{3} & 0 & \frac{11}{6} & -\frac{11}{6} & -\frac{47}{10} & z1 \\ 0 & 0 & -\frac{5}{3} & 0 & -\frac{5}{3} & \frac{8}{3} & -\frac{1}{2} & z2 \end{pmatrix}$$

(%i345) b2ratio (RL, 3);

(%o345)
$$\begin{pmatrix} 1.2 & x1 \\ 0.3 & x4 \\ - & x2 \end{pmatrix}$$

(%i346) RL : pivot1 (RL, [2, 3])$

$$pivot\ row = 2 \quad pivot\ col = 3 \quad value = \frac{5}{3}$$

x3 enters Basis, x4 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & x4 & x5 & x6 & rhs & Basis \\ 1 & 0 & 0 & -4 & -5 & 5 & 6 & x1 \\ 0 & 0 & 1 & \frac{3}{5} & 1 & -1 & \frac{3}{10} & x3 \\ 0 & 1 & 0 & 6 & 5 & -5 & 6 & x2 \\ 0 & 0 & 0 & -\frac{7}{5} & -\frac{1}{2} & \frac{1}{2} & -\frac{27}{5} & z1 \\ 0 & 0 & 0 & 1 & 0 & 1 & 0 & z2 \end{pmatrix}$$

Artificial variables x4 and x6 have left the Basis and are equal to zero. Remove columns 4 and 6 and preserve the remaining columns. Remove the last row.

```
(%i347) r1 : remL (RL[1], [4, 6]);
```

(r1) $[1,0,0,-5,6]$

```
(%i348) r2 : remL (RL[2], [4, 6]);
```

(r2) $[0,0,1,1,\frac{3}{10}]$

```
(%i349) r3 : remL (RL[3], [4, 6]);
```

(r3) $[0,1,0,5,6]$

```
(%i350) r4 : remL (RL[4], [4, 6]);
```

(r4) $[0,0,0,-\frac{1}{2},-\frac{27}{5}]$

```
(%i354) vL : [x1, x2, x3, x5, rhs, Basis]$
bL : [x1, x3, x2, z]$
RL : [r1,r2,r3,r4]$
tableau (RL);
```

(%o354)

x1	x2	x3	x5	rhs	Basis
1	0	0	−5	6	x1
0	0	1	1	$\frac{3}{10}$	x3
0	1	0	5	6	x2
0	0	0	$-\frac{1}{2}$	$-\frac{27}{5}$	z

```
(%i355) bratio (RL, 4);
```

(%o355)

−	x1
0.3	x3
1.2	x2

(%i356) RL : pivot1 (RL, [2, 4])$

pivot row = 2 *pivot col =* 4 *value =* 1

x5 enters Basis, x3 leaves Basis

$$\begin{pmatrix} x1 & x2 & x3 & x5 & rhs & Basis \\ 1 & 0 & 5 & 0 & \frac{15}{2} & x1 \\ 0 & 0 & 1 & 1 & \frac{3}{10} & x5 \\ 0 & 1 & -5 & 0 & \frac{9}{2} & x2 \\ 0 & 0 & \frac{1}{2} & 0 & -\frac{21}{4} & z \end{pmatrix}$$

No more negative coefficients in the z-row (omitting current value of z), so a primal optimum solution is z* = -w* = - 21/4 = -5.25, w* = 21/4 = 5.25, x1* = 15/2 = 7.5, x2* = 9/2 = 4.5, which agrees with the solution found by minimize_lp.

The Basis variables in the optimum tableau are (x1,x2,x5) which are elements (1,2,5) of Xs. Let Cb be elements (1,2,5) of Cs and Ab be a matrix formed from columns (1,2,5) of the matrix As.

(%i357) Cs;

(%o357) $$\begin{pmatrix} 0.4 \\ 0.5 \\ 0 \\ M \\ 0 \\ M \end{pmatrix}$$

(%i358) Cb : part (Cs, [1,2,5]);

(Cb) $$\begin{pmatrix} 0.4 \\ 0.5 \\ 0 \end{pmatrix}$$

(%i359) As;

(%o359) $$\begin{pmatrix} 0.3 & 0.1 & 1 & 0 & 0 & 0 \\ 0.5 & 0.5 & 0 & 1 & 0 & 0 \\ 0.6 & 0.4 & 0 & 0 & -1 & 1 \end{pmatrix}$$

(%i360) Ab : newM (As, [1,2,5]);

$$(Ab)\quad \begin{pmatrix} 0.3 & 0.1 & 0 \\ 0.5 & 0.5 & 0 \\ 0.6 & 0.4 & -1 \end{pmatrix}$$

(%i361) transpose (Cb) . invert (Ab);

$$(\%o361)\quad \begin{pmatrix} -0.5 & 1.1 & 0.0 \end{pmatrix}$$

This predicts a solution of the dual LP is y1 = -0.5, y2 = 1.1, y3 = 0.0, with
z = -2.7*y1 + 6*y2 + 6*y3 = 7.95. This prediction of the dual LP solution does not satisfy the dual LP conditions y1, y3 >= 0, y2 unrestricted in sign (urs).

6.2.3 Simplex Tableau Solution to Dual LP

The dual LP is:
maximize z = -2.7*y1 + 6*y2 + 6*y3,
subject to
-0.3*y1 + 0.5*y2 + 0.6*y3 <= 0.4,
-0.1*y1 + 0.5*y2 + 0.4*y3 <= 0.5,
with y1, y3 >= 0, and y2 unrestricted in sign (urs).

(%i362) maximize_lp (-2.7*y1 + 6*y2 + 6*y3, [-0.3*y1 + 0.5*y2 + 0.6*y3 <= 0.4,
-0.1*y1 + 0.5*y2 + 0.4*y3 <= 0.5], [y1,y3]);

Warning: linear_program(A,b,c): non-rat inputs found, epsilon_lp=1.e-8.

Warning: Solution may be incorrect.

$$(\%o362)\quad [5.25,[y3=0,y2=1.1,y1=0.5]]$$

Since y2 is unrestricted in sign, set y2 = y4 - y5 with y4,y5 >= 0.
Add slack y6 to left-hand side of first constraint, slack y7 to left-hand side of second constraint to convert constraints to equations.
We then have Y^t = (y1,y3,y4,y5,y6,y7), max z = u^t . Y with u^t = (-2.7, 6, 6, -6, 0, 0) subject to D . Y = v, with Y >= 0. v^t = (0.4, 0.5), initial feasible solution:
y1 = y3 = y4 = y5 = 0, y6 = 0.4, y7 = 0.5, Yo^t = (y6,y7), uo^t = (0,0)

Given the Step 0 LP: maximize z = u^t . Y, such that D . Y = v, with Y >= 0,
the step 0 maximization tableau is, using these matrices:

Y^t	rhs	Basis
D	v	Xso
uo^t . D - u^t	uo^t . v	z

```
(%i372) u : cvec ([-2.7, 6, 6, -6, 0, 0])$
uo : cvec ([0,0])$
D : matrix ( [-0.3, 0.6, 0.5, -0.5, 1, 0], [-0.1, 0.4, 0.5, -0.5, 0, 1])$
Y : cvec ([y1,y3,y4,y5,y6,y7])$
v : cvec ([0.4, 0.5])$
"maximize z" = transpose(u) . Y;
"subject to";
D . Y  = v;
"uo^t . D - u^t " = transpose(uo) . D - transpose(u);
"uo^t . v " = transpose (uo) . v;
```

(%o368) *maximize* $z = -6\,y5 + 6\,y4 + 6\,y3 - 2.7\,y1$

(%o369) *subject to*

(%o370) $\begin{pmatrix} y6 - 0.5\,y5 + 0.5\,y4 + 0.6\,y3 - 0.3\,y1 \\ y7 - 0.5\,y5 + 0.5\,y4 + 0.4\,y3 - 0.1\,y1 \end{pmatrix} = \begin{pmatrix} 0.4 \\ 0.5 \end{pmatrix}$

(%o371) *uo^t . D − u^t* $= \begin{pmatrix} 2.7 & -6.0 & -6.0 & 6.0 & 0 & 0 \end{pmatrix}$

(%o372) *uo^t . v* $= 0.0$

```
(%i379) vL : [y1,y3,y4,y5,y6,y7, rhs, Basis]$
bL : [y6,y7,z]$
r1 : endcons (v[1,1], D[1])$
r2 : endcons (v[2,1], D[2])$
r3 : [2.7,-6,-6,6,0,0, 0]$
RL : [r1,r2,r3]$
tableau (RL);
```

(%o379)

y1	y3	y4	y5	y6	y7	rhs	Basis
−0.3	0.6	0.5	−0.5	1	0	0.4	y6
−0.1	0.4	0.5	−0.5	0	1	0.5	y7
2.7	−6	−6	6	0	0	0	z

```
(%i380) bratio(RL, 2);
```

(%o380)

0.66667	y6
1.25	y7

(%i381) RL : pivot1 (RL, [1, 2])$

pivot row = 1 *pivot col* = 2 *value* = 0.6

y3 enters Basis, y6 leaves Basis

$$\begin{pmatrix} y1 & y3 & y4 & y5 & y6 & y7 & rhs & Basis \\ -\frac{1}{2} & 1 & \frac{5}{6} & -\frac{5}{6} & \frac{5}{3} & 0 & \frac{2}{3} & y3 \\ \frac{1}{10} & 0 & \frac{1}{6} & -\frac{1}{6} & -\frac{2}{3} & 1 & \frac{7}{30} & y7 \\ -\frac{3}{10} & 0 & -1 & 1 & 10 & 0 & 4 & z \end{pmatrix}$$

(%i382) bratio (RL, 3);

(%o382) $\begin{pmatrix} 0.8 & y3 \\ 1.4 & y7 \end{pmatrix}$

(%i383) RL : pivot1 (RL, [1,3])$

pivot row = 1 *pivot col* = 3 *value* = $\frac{5}{6}$

y4 enters Basis, y3 leaves Basis

$$\begin{pmatrix} y1 & y3 & y4 & y5 & y6 & y7 & rhs & Basis \\ -\frac{3}{5} & \frac{6}{5} & 1 & -1 & 2 & 0 & \frac{4}{5} & y4 \\ \frac{1}{5} & -\frac{1}{5} & 0 & 0 & -1 & 1 & \frac{1}{10} & y7 \\ -\frac{9}{10} & \frac{6}{5} & 0 & 0 & 12 & 0 & \frac{24}{5} & z \end{pmatrix}$$

(%i384) bratio (RL, 1);

(%o384) $\begin{pmatrix} - & y4 \\ 0.5 & y7 \end{pmatrix}$

(%i385) RL : pivot1 (RL, [2,1])$

$$pivot\ row =\ 2\quad pivot\ col =\ 1\quad value = \frac{1}{5}$$

y1 enters Basis, y7 leaves Basis

$$\begin{pmatrix} y1 & y3 & y4 & y5 & y6 & y7 & rhs & Basis \\ 0 & \frac{3}{5} & 1 & -1 & -1 & 3 & \frac{11}{10} & y4 \\ 1 & -1 & 0 & 0 & -5 & 5 & \frac{1}{2} & y1 \\ 0 & \frac{3}{10} & 0 & 0 & \frac{15}{2} & \frac{9}{2} & \frac{21}{4} & z \end{pmatrix}$$

Optimum dual LP tableau with z = 21/4 = 5.25, y1 = 1/5 = 0.5, y2 = y4 - y5 = 11/10 = 1.1, y3 = 0.0, agrees with maximize_lp solution.

```
(%i388) ub : part (u, [1,3])$
        Db : newM (D, [1,3])$
        transpose (ub) . invert (Db);
```

(%o388) $\begin{pmatrix} 7.5 & 4.5 \end{pmatrix}$

This agrees with the optimal primal LP solution found x1 = 7.5, x2 = 4.5, and satisfies the primal conditions x1,x2 >= 0.

7 Appendix: Step 0 Tableaux

Xso is the known initial feasible solution Basis vector of symbols.
Xso is defined using the Basis variable order in the constraint equations, and not necessarily with the order in Xs.
Xso has the same number of variables as the rhs vector E.
Cso is the vector of objective coefficients, taken from Cs, associated with the initial basis vector Xso, and in the same order as Xso.

Given the Step 0 LP: minimize w = Cs^t . Xs, such that As . Xs = E, with Xs >= 0,

the step 0 minimization tableau is, using these matrices:

```
   |        Xs^t           |   rhs         |  Basis
-----------------------------------------------------------
   |         As            |   E           |  Xso
-----------------------------------------------------------
   |  Cs^t - Cso^t . As    | - Cso^t . E   |  z
```

Given the Step 0 LP: maximize z = Cs^t . Xs, such that As . Xs = E, with Xs >= 0,

the step 0 maximization tableau is, using these matrices:

Xs^t	rhs	Basis
As	E	Xso
Cso^t . As - Cs^t	Cso^t . E	z